Cinematica diretta Robot sferico I tipo

N.B.: **le grandezze diverse da quelle di giunto q_i sono L_i, D_i. Esse sono rispettivamente la distanza tra i sistemi di riferimento R_i e R_{i+1} nelle operazioni della matrice avvitamento $A_z(\theta, d)$ e $A_x(\alpha, a)$.**



	ϑ	d	α	a
1	q_1	L_1	$\frac{\pi}{2}$	0
2	q_2	0	$\frac{\pi}{2}$	L_2
3	0	q_3	0	0

Tabella 1.

Funzioni ausiliarie:

```
(%i1) inverseLaplace(SI,theta):=block([res],
                            M:SI,
                            MC:SI,
                            for i:1 thru 3 do(
                              for j:1 thru 3 do
                                 (
                                   aC:M[i,j],
                                   b:ilt(aC,s,theta),
                                   MC[i,j]:b
                                  )
                               ),
                            res:MC

                         )
```

(%o1) inverseLaplace(SI, ϑ) := **block** ([res], M: SI, MC: SI, **for** i **thru** 3 **do for** j **thru** 3 **do** (aC:

$M_{i,j}, b$: ilt(aC, s, ϑ), MC$_{i,j}$: b), res: MC)

(%i2)
```
rotLaplace(k,theta):=block([res],
                      S:ident(3),
                      I:ident(3),
                    for i:1 thru 3 do
                      (
                      for j:1 thru 3 do
                         (
                           if i=j
                              then S[i][j]:0
                           elseif j>i
                              then (
                             temp:(-1)^(j-i)*k[3-remainder(i+j,3)],
                                    S[i][j]:temp,
                                    S[j][i]:-temp

                                     )
                           )
                       ),
                      res:inverseLaplace(invert(s*I-S),theta)


                     )
```

(%o2) $\mathrm{rotLaplace}(k,\vartheta) := \mathbf{block}$ ([res], S: ident(3), I: ident(3), **for** i **thru** 3 **do for** j **thru** 3 **do if** $i=j$ **then** $(S_i)_j$: 0 **elseif** $j>i$ **then** (temp: $(-1)^{j-i}\, k_{3-\mathrm{remainder}(i+j,3)}$, $(S_i)_j$: temp, $(S_j)_i$: $-$temp), res: inverseLaplace(invert($s\,I-S$), ϑ))

(%i3)
```
Av(v,theta,d):=block([res],
                      Trot:rotLaplace(v,theta),
                      row:matrix([0,0,0,1]),
                      Atemp:addcol(Trot,d*transpose(v)),
                      A:addrow(Atemp,row),
                      res:A
                      )
```

(%o3) $\mathrm{Av}(v,\vartheta,d) := \mathbf{block}$ ([res], Trot: rotLaplace(v, ϑ), row: $\begin{pmatrix} 0 & 0 & 0 & 1 \end{pmatrix}$, Atemp: addcol(Trot, d transpose(v)), A: addrow(Atemp, row), res: A)

(%i4)
```
Q(theta,d,alpha,a):=block([res],
                           tempMat:Av([0,0,1],theta,d).Av([1,0,0],alpha,a),
                           Qtrasf:zeromatrix(4,4),
                           for i:1 thru 4 do
                      (
                      for j:1 thru 4 do
                         (
                           Qtrasf[i][j]:trigreduce(tempMat[i][j])
                                     )
                         ),

                           res:Qtrasf
                           )
```

(%o4) $Q(\vartheta,d,\alpha,a) := \mathbf{block}$ ([res], tempMat: Av([0, 0, 1], ϑ, d) $\cdot$ Av([1, 0, 0], α, a), Qtrasf:

zeromatrix$(4,4)$, **for** i **thru** 4 **do for** j **thru** 4 **do** $(\mathrm{Qtrasf}_i)_j$: trigreduce$((\mathrm{tempMat}_i)_j)$, res: Qtrasf)

(%i5) `let(sin(q[1]), s[1]);`

(%o5) $\sin(q_1) \longrightarrow s_1$

(%i6) `let(sin(q[2]), s[2]);`

(%o6) $\sin(q_2) \longrightarrow s_2$

(%i7) `let(cos(q[1]),c[1]);`

(%o7) $\cos(q_1) \longrightarrow c_1$

(%i8) `let(cos(q[2]),c[2]);`

(%o8) $\cos(q_2) \longrightarrow c_2$

(%i9) `let(sin(q[1]+q[2]), s[12]);`

(%o9) $\sin(q_2+q_1) \longrightarrow s_{12}$

(%i10) `let(cos(q[1]+q[2]),c[12]);`

(%o10) $\cos(q_2+q_1) \longrightarrow c_{12}$

(%i11)

Cinematica diretta:

(%i11)
```
Q[sfericoRRP](q1,q2,q3,L1,L2):=
                  Q(q1,L1,%pi/2,0).
                  Q(q2,0,%pi/2,L2).
                  Q(0,q3,0,0)
                  ;
```

(%o11) $Q_{\mathrm{sfericoRRP}}(q1,q2,q3,L1,L2) := Q\left(q1, L1, \frac{\pi}{2}, 0\right) \cdot Q\left(q2, 0, \frac{\pi}{2}, L2\right) \cdot Q(0,q3,0,0)$

(%i12) `QsfericoRRP:Q[sfericoRRP](q[1],q[2],q[3],L[1],L[2]);`

(%o12)
$$\begin{pmatrix} \cos(q_1)\cos(q_2) & \sin(q_1) & \cos(q_1)\sin(q_2) & \cos(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2)) \\ \sin(q_1)\cos(q_2) & -\cos(q_1) & \sin(q_1)\sin(q_2) & \sin(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2)) \\ \sin(q_2) & 0 & -\cos(q_2) & L_2\sin(q_2)-q_3\cos(q_2)+L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

(%i13) `letsimp(QsfericoRRP);`

(%o13)
$$\begin{pmatrix} c_1 c_2 & s_1 & c_1 s_2 & c_1 s_2 q_3 + c_1 L_2 c_2 \\ s_1 c_2 & -c_1 & s_1 s_2 & s_1 s_2 q_3 + s_1 L_2 c_2 \\ s_2 & 0 & -c_2 & -c_2 q_3 + L_2 s_2 + L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

(%i14)

Cinematica inversa robot sferico I tipo

Al fine di risolvere il problema di cinematica inversa del robot sferico di I tipo occorre risolevere il problema di posizione ed orientamente inverso.Inizialmente individuare lo spazio di lavoro, le soluzioni generiche, singolari ed infine le variabili di giunto q_i ed in seguito determinare l'orientamento del robot.

Dalla cinematica diretta del robot sferico di I tipo, sappiamo che:

$$Q_{\text{sferico}} = \begin{pmatrix} c_1\, c_2 & s_1 & c_1\, s_2 & c_1\, s_2\, q_3 + c_1\, L_2\, c_2 \\ s_1\, c_2 & -c_1 & s_1\, s_2 & s_1\, s_2\, q_3 + s_1\, L_2\, c_2 \\ s_2 & 0 & -c_2 & -c_2\, q_3 + L_2\, s_2 + L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Cinematica inversa di posizione

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} c_1\, s_2\, q_3 + c_1\, L_2\, c_2 \\ s_1\, s_2\, q_3 + s_1\, L_2\, c_2 \\ -c_2\, q_3 + L_2\, s_2 + L_1 \end{pmatrix}$$

$$\begin{cases} x = c_1\, s_2\, q_3 + c_1\, L_2\, c_2 \\ y = s_1\, s_2\, q_3 + s_1\, L_2\, c_2 \end{cases} \longrightarrow \begin{cases} x^2 = c_1^2\, (s_2\, q_3 + L_2\, c_2)^2 \\ y = s_1^2 (s_2\, q_3 + L_2\, c_2)^2 \end{cases}$$

$$x^2 + y^2 = (c_1^2 + s_1^2)\, (s_2\, q_3 + L_2\, c_2)^2 \rightarrow \begin{cases} x^2 + y^2 = (c_1^2 + s_1^2)\, (s_2\, q_3 + L_2\, c_2)^2 \\ z = -c_2\, q_3 + L_2\, s_2 + L_1 \end{cases}$$

$$\begin{cases} L_1 - z = -c_2\, q_3 + L_2\, s_2 \\ q_3 s_2 + L_2 c_2 = \pm\sqrt{x^2 + y^2} \end{cases}$$

A questo punto è possibile riscrivere l'ultima espressione come:

$$\begin{pmatrix} c_2 & -s_2 \\ s_2 & c_2 \end{pmatrix} \begin{pmatrix} q_3 \\ L_2 \end{pmatrix} = \begin{pmatrix} L_1 - z \\ \pm\sqrt{x^2 + y^2} \end{pmatrix}$$

Poiché $\begin{pmatrix} c_2 & -s_2 \\ s_2 & c_2 \end{pmatrix}$ è una matrice di rotazione, *i* vettori $\begin{pmatrix} q_3 \\ L_2 \end{pmatrix}$ e $\begin{pmatrix} L_1 - z \\ \pm\sqrt{x^2 + y^2} \end{pmatrix}$ devono avere la stessa norma.Imponendo questa condizione, si ottiene l'espressione dello spazio operativo:

$$q_3^2 + L_2^2 = (L_1 - z)^2 + x^2 + y^2$$

Nell'ultima equazione è presente una variabile di giunto che, una volta fissata, permette la definizione del luogo geometrico che identifica lo spazio operativo. In definitiva, il luogo geometrico descritto è una sfera cava di centro $\begin{pmatrix} 0 \\ 0 \\ L_1 \end{pmatrix}$ e raggio $r_{q_3} = \sqrt{q_3^2 + L_2^2}$ di cui si ottiene il raggio interno imponendo $q_3 = 0 \rightarrow r = L_2$ e il raggio esterno per $q_3 \rightarrow \pm\infty\ \rightarrow r = +\infty$.

Una volta determinato lo spazio operativo, occorre determinare i punti di singolarità e le soluzioni generiche:

$$q_3 = \pm\sqrt{x^2 + y^2 + (z - L_1)^2 - L_2^2}$$

In particolare, si hanno due soluzioni generiche ed una sincolarita se:

$$x^2 + y^2 + (z - L_1)^2 = L_2^2$$

che coincide con la sfera di raggio più piccolo.

Inoltre, definendo l'espressione di q_2, L_2 e di $L_1 - z, \pm\sqrt{x^2+y^2}$ come rispettivamente A_1, A_2, B_1, B_2, possiamo calcolare le altre 2 variabili di giunto:

$$\begin{cases} A_1 c_2 - A_2 s_2 = B_1 \\ A_1 s_2 + A_2 c_2 = B_2 \end{cases} \longrightarrow \begin{pmatrix} A_1 & -A_2 \\ A_2 & A_1 \end{pmatrix} \begin{pmatrix} c_2 \\ s_2 \end{pmatrix} = \begin{pmatrix} B_1 \\ B_2 \end{pmatrix}$$

Il determinante della matrice $\det\begin{pmatrix} A_1 & -A_2 \\ A_2 & A_1 \end{pmatrix} \neq 0$ nei punti non singolari e, sotto questa ipotesi è possibile effetuare l'inversa:

$$\begin{pmatrix} c_2 \\ s_2 \end{pmatrix} = \frac{1}{A_1^2 + A_2^2} \begin{pmatrix} A_1 & A_2 \\ -A_2 & A_1 \end{pmatrix} \begin{pmatrix} B_1 \\ B_2 \end{pmatrix}$$

Quindi la variabile di giunto q_2 ha la seguente espressione:

$$q_2 = \operatorname{atan2}(-A_2 B_1 + A_1 B_2, A_1 B_1 + A_2 B_2)$$

A questo punto la quantità $(s_2\, q_3 + L_2\, c_2)$ è nota e supponiamo che sia $\neq 0$:

$$c_1 = \frac{x}{s_2\, q_3 + L_2\, c_2} = \frac{x}{\sqrt{x^2+y^2}}$$
$$s_1 = \frac{y}{s_2\, q_3 + L_2\, c_2} = \frac{y}{\sqrt{x^2+y^2}}$$

Quindi:

$$q_1 = \operatorname{atan2}\left(\frac{y}{\sqrt{x^2+y^2}}, \frac{x}{\sqrt{x^2+y^2}} \right)$$

Dato che $\sqrt{x^2+y^2} > 0$:

$$q_1 = \operatorname{atan2}(y, x)$$

Orientamento inverso

La risoluzione del problema di orientamento inverso si basa sulla scelta di una terna di Eulero o nautica in condizione non singolari, se possibile.

$$R_{\text{sfericoI}} = \begin{pmatrix} \cos(q_1)\cos(q_2) & \sin(q_1) & \cos(q_1)\sin(q_2) \\ \sin(q_1)\cos(q_2) & -\cos(q_1) & \sin(q_1)\sin(q_2) \\ \sin(q_2) & 0 & -\cos(q_2) \end{pmatrix}$$

$$R_{\text{zyx}} = \begin{pmatrix} c_y c_z & \cdots\cdots\cdots & \\ c_y s_z & \cdots\cdots\cdots & \\ -s_y & s_x c_y & c_x c_y \end{pmatrix}$$

Uguagliando i termini $R_{3,1}$ della matrice di rotazione R_{sfericoI} con quella della matrice di rotazione con angoli nautivi R_{zyx}:

$$-s_y = \sin(q_2)\,\text{singolare se} -s_y = \sin(q_2) = \pm 1 \rightarrow q_2 = \pm\frac{\pi}{2}$$

Supponsiamo che $q_2 \neq \pm\frac{\pi}{2}$:

$$\begin{cases} s_y = -\sin(q_2) \\ c_y = \pm\sqrt{1-s_2^2} = \pm c_2 \end{cases} \longrightarrow \phi_y = \text{atan2}(-s_2 \pm c_2) = \begin{cases} -q_2 \\ q_2 + \pi \end{cases}$$

Inoltre:

$$\begin{cases} s_x c_y = 0 \\ c_x c_y = \mp 1 \end{cases} \longrightarrow \begin{cases} \pm s_x c_2 = 0 \\ \pm c_x c_2 = -c_2 \end{cases} \longrightarrow \begin{cases} s_x = 0 \\ c_x = \mp 1 \end{cases} \rightarrow \phi_x = \text{atan2}(0, \mp 1) = \begin{cases} \pi \\ 0 \end{cases}$$

$$\begin{cases} c_y c_z = c_1 c_2 \\ c_y c_z = s_1 c_2 \end{cases} \rightarrow \begin{cases} \pm c_2 c_z = c_1 c_2 \\ \pm c_2 c_z = s_1 c_2 \end{cases} \rightarrow \begin{cases} c_z = \pm c_1 \\ s_z = \pm s_1 \end{cases} \rightarrow \phi_z = \text{atan2}(\pm s_1, c_1) = \begin{cases} q_1 \\ q_1 + \pi \end{cases}$$

Riassumendo:

$$\begin{pmatrix} \pi \\ -q_2 \\ q_1 \end{pmatrix}, \begin{pmatrix} 0 \\ q_2 + \pi \\ q_1 + \pi \end{pmatrix}$$

In alternativa tramite una scelta di una terna di eulero:

$$R_{\text{zyz}} = \begin{pmatrix} \cdots & \cdots & \cos(\alpha)\sin(\beta) \\ \cdots & \cdots & \sin(\alpha)\sin(\beta) \\ -\sin(\beta)\cos(\gamma) & \sin(\beta)\sin(\gamma) & \cos(\beta) \end{pmatrix}$$

$$\cos(\beta) = -\cos(q_2) \neq \pm 1 \rightarrow q_2 \neq \begin{cases} \pi \\ 0 \end{cases}$$

Supponiamo che $q_2 \neq \begin{cases} \pi \\ 0 \end{cases}$ per non avere una soluzione singolare:

$$\sin(\beta) = \pm\sqrt{1-\cos(\beta)^2} = \pm\sqrt{1-\cos(q_2)^2} = \pm\sin(q_2)$$

$$\beta = \text{atan2}(\pm\sin(q_2), \cos(q_2)) = \begin{cases} q_2 \\ -q_2 \end{cases}$$

$$\begin{cases} \cos(\alpha)\sin(\beta) = c_1 s_2 \\ \sin(\alpha)\sin(\beta) = s_1 s_2 \end{cases} \rightarrow \begin{cases} \pm\cos(\alpha)\sin(q_2) = \cos(q_1)\sin(q_2) \\ \pm\sin(\alpha)\sin(q_2) = \sin(q_1)\sin(q_2) \end{cases} \rightarrow \begin{cases} \cos(\alpha) = \pm\cos(q_1) \\ \sin(\alpha) = \pm\sin(q_1) \end{cases}$$

$$\alpha = \text{atan2}(\pm\sin(q_1), \pm\cos(q_1)) = \begin{cases} q_1 \\ q_1 + \pi \end{cases}$$

$$\begin{cases} -\sin(\beta)\cos(\gamma) = s_2 \\ \sin(\beta)\sin(\gamma) = 0 \end{cases} \rightarrow \begin{cases} \mp\cos(\gamma) = + \\ \sin(\gamma) = 0 \end{cases} \longrightarrow \gamma = \text{atan2}(0, \mp 1) = \begin{cases} \pi \\ 0 \end{cases}$$

Riassumendo:

$$\begin{pmatrix} q_1 \\ q_2 \\ \pi \end{pmatrix}, \begin{pmatrix} q_1 + \pi \\ -q_2 \\ 0 \end{pmatrix}$$

```
(%i14) isRotation(M):=block([MC,res],
                          I:ident(3),
                          MC:ident(3),
                          for i:1 thru 3 do
                          (
                          for j:1 thru 3 do
                             (
                                MC[i][j]:M[i][j]
                                )
                             ),
                          MMT:trigsimp(expand(MC.transpose(MC))),
                          detM:trigsimp(expand(determinant(MC))),

                          if MMT=I  and detM=1
                             then(

                                  return(res:1)
                                  )

                          else(

                                res: "R is not rotation matrix"
                                )
                          )
```

(%o14) isRotation(M) := **block** ([MC, res], I: ident(3), MC: ident(3), **for** i **thru** 3 **do for** j **thru** 3 **do** $(\mathrm{MC}_i)_j$: $(M_i)_j$, MMT: trigsimp(expand(MC $\cdot$ transpose(MC))), detM: trigsimp(expand(determinant(MC))), **if** MMT $= I \wedge$ detM $= 1$ **then** return(res: 1) **else** res: R is not rotation matrix)

```
(%i15) calculate(x,y,L1,L2,z):=block([q3plus,q3minus,q2plus,q2minus,q1,res],

             if x^(2)+y^(2)+(z-L1)^2=L2^(2) then print("La soluzione è singolare")
                                 else
                                 (
       q3value:sqrt(trigreduce(trigexpand(ratsimp(x^(2)+y^(2)+(z-L1)^(2)-
       L2^(2))))),
       q3plus:trigreduce(trigexpand(ratsimp(q3value))),
       q3minus:-trigreduce(trigexpand(ratsimp(q3value))),
       twoNorm:trigreduce(trigexpand(sqrt(x^(2)+y^(2)))),
       s2plus:trigreduce(trigexpand(ratsimp(-L2*(L1-z)+q3plus*twoNorm))),
       c2plus:trigreduce(trigexpand(ratsimp(q3plus*(L1-z)+L2*twoNorm))),
       s2minus:trigreduce(trigexpand(ratsimp(-L2*(L1-z)-q3minus*twoNorm))),
       c2minus:trigreduce(trigexpand(ratsimp(q3minus*(L1-z)-L2*twoNorm))),
                                  q2plus:atan2(s2plus,c2plus),
                                  q2minus:atan2(s2minus,c2minus),
                                  q1:atan2(y,x),
                                  res:[[q1,q2plus,q3plus],[q1,q2minus,q3minus]]
                                 )
                                 )
```

(%o15) $\mathrm{calculate}(x,\, y,\, L1,\, L2,\, z) := \mathbf{block}\left([q3\mathrm{plus},\, q3\mathrm{minus},\, q2\mathrm{plus},\, q2\mathrm{minus},\, q1,\, \mathrm{res}],\, \mathbf{if}\ x^2 + y^2 + (z - L1)^2 = L2^2\ \mathbf{then}\ \mathrm{print}(\text{La soluzione è singolare })\ \mathbf{else}\left(q3\mathrm{value}\colon \sqrt{\mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{ratsimp}(x^2 + y^2 + (z - L1)^2 - L2^2)))},\, q3\mathrm{plus}\colon \mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{ratsimp}(q3\mathrm{value}))),\, q3\mathrm{minus}\colon -\mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{ratsimp}(q3\mathrm{value}))),\, \mathrm{twoNorm}\colon \mathrm{trigreduce}\left(\mathrm{trigexpand}\left(\sqrt{x^2+y^2}\right)\right),\, s2\mathrm{plus}\colon \mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{ratsimp}((-L2)\,(L1 - z) + q3\mathrm{plus}\,\mathrm{twoNorm}))),\, c2\mathrm{plus}\colon \mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{ratsimp}(q3\mathrm{plus}\,(L1 - z) + L2\,\mathrm{twoNorm}))),\, s2\mathrm{minus}\colon \mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{ratsimp}((-L2)\,(L1 - z) - q3\mathrm{minus}\,\mathrm{twoNorm}))),\, c2\mathrm{minus}\colon \mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{ratsimp}(q3\mathrm{minus}\,(L1 - z) - L2\,\mathrm{twoNorm}))),\, q2\mathrm{plus}\colon \mathrm{atan2}(s2\mathrm{plus},\, c2\mathrm{plus}),\, q2\mathrm{minus}\colon \mathrm{atan2}(s2\mathrm{minus},\, c2\mathrm{minus}),\, q1\colon \mathrm{atan2}(y, x),\, \mathrm{res}\colon [[q1,\, q2\mathrm{plus},\, q3\mathrm{plus}],\, [q1,\, q2\mathrm{minus},\, q3\mathrm{minus}]]\right)\right)$

(%i28)
```
orientation(Qdiretta):=block([sx,cx,sy,cy,phiy1,phiy2,phiz1,phiz2,phix1,
phix2,sz,sxfirst,second,res],

                              rotation:isRotation(Qdiretta),
                              if rotation=1 then(
                              sy:Qdiretta[3][1],

                              if sy=1 or sy=-1 then "soluzione singolare"
                              else(
                              cy:sqrt(1-sy^2),
                              phiy1:atan2(-sy,cy),
                              phiy2:atan2(-sy,-cy),

                              sx:Qdiretta[3][2]/cy,
                              cx:Qdiretta[3][3]/cy,
                              phix1:atan2(sx,cx),
                              phix2:atan2(sx,-cx),
                              cz:Qdiretta[1][1]/cy,
                              sz:Qdiretta[2][1]/cy,
                              phiz1:atan2(sz,cz),
                              phiz2:atan2(-sz,-cz),
                              first:[phix1,phiy1,phiz1],
                              second:[phix2,phiy2,phiz2],

                              res:[first,second])
                              )
                              else res:rotation

);
```

(%o28) $\mathrm{orientation}(\mathrm{Qdiretta}) := \mathbf{block}\left([\mathrm{sx},\, \mathrm{cx},\, \mathrm{sy},\, \mathrm{cy},\, \mathrm{phiy1},\, \mathrm{phiy2},\, \mathrm{phiz1},\, \mathrm{phiz2},\, \mathrm{phix1},\, \mathrm{phix2},\, \mathrm{sz},\, \mathrm{sxfirst},\, \mathrm{second},\, \mathrm{res}],\, \mathrm{rotation}\colon \mathrm{isRotation}(\mathrm{Qdiretta}),\, \mathbf{if}\ \mathrm{rotation} = 1\ \mathbf{then}\left(\mathrm{sy}\colon (\mathrm{Qdiretta}_3)_1,\, \mathbf{if}\ \mathrm{sy} = 1 \lor \mathrm{sy} = -1\ \mathbf{then}\ \text{soluzione singolare }\ \mathbf{else}\left(\mathrm{cy}\colon \sqrt{1 - \mathrm{sy}^2},\, \mathrm{phiy1}\colon \mathrm{atan2}(-\mathrm{sy},\, \mathrm{cy}),\, \mathrm{phiy2}\colon \mathrm{atan2}(-\mathrm{sy},\, -\mathrm{cy}),\, \mathrm{sx}\colon \frac{(\mathrm{Qdiretta}_3)_2}{\mathrm{cy}},\, \mathrm{cx}\colon \frac{(\mathrm{Qdiretta}_3)_3}{\mathrm{cy}},\, \mathrm{phix1}\colon \mathrm{atan2}(\mathrm{sx},\, \mathrm{cx}),\, \mathrm{phix2}\colon \mathrm{atan2}(\mathrm{sx},\, -\mathrm{cx}),\, \mathrm{cz}\colon \frac{(\mathrm{Qdiretta}_1)_1}{\mathrm{cy}},\, \mathrm{sz}\colon \frac{(\mathrm{Qdiretta}_2)_1}{\mathrm{cy}},\, \mathrm{phiz1}\colon \mathrm{atan2}(\mathrm{sz},\, \mathrm{cz}),\, \mathrm{phiz2}\colon \mathrm{atan2}(-\mathrm{sz},\, -\mathrm{cz}),\, \mathrm{first}\colon [\mathrm{phix1},\, \mathrm{phiy1},\, \mathrm{phiz1}],\, \mathrm{second}\colon [\mathrm{phix2},\, \mathrm{phiy2},\, \mathrm{phiz2}],\, \mathrm{res}\colon [\mathrm{first},\, \mathrm{second}]\right)\right)\ \mathbf{else}\ \mathrm{res}\colon \mathrm{rotation}\right)$

```
(%i29) invSfericoI(Qdiretta,L1,L2):=block([x,y,z,pos1,pos2,orien1,orien2,res],
                                         x:Qdiretta[1][4],
                                         y:Qdiretta[2][4],
                                         z:Qdiretta[3][4],
                                         pos:calculate(x,y,L1,L2,z),
                                         orien:orientation(Qdiretta),
                                         orien1:orien[1],
                                         orien2:orien[2],
                                         pos1:pos[1],
                                         pos2:pos[2],
                                         res:[pos1,pos2,orien1,orien2]
                                         );
```

(%o29) invSfericoI(Qdiretta, $L1$, $L2$) := **block** ([x, y, z, pos1, pos2, orien1, orien2, res], x: $(\mathrm{Qdiretta}_1)_4$, y: $(\mathrm{Qdiretta}_2)_4$, z: $(\mathrm{Qdiretta}_3)_4$, pos: calculate(x, y, $L1$, $L2$, z), orien: orientation(Qdiretta), orien1: orien_1, orien2: orien_2, pos1: pos_1, pos2: pos_2, res: [pos1, pos2, orien1, orien2])

```
(%i30) QsfericoRRP:Q[sferico](%pi/3,0,10,5,10);
```

(%o30)
$$\begin{pmatrix} \frac{1}{2} & \frac{\sqrt{3}}{2} & 0 & 5 \\ \frac{\sqrt{3}}{2} & -\frac{1}{2} & 0 & 5\sqrt{3} \\ 0 & 0 & -1 & -5 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(%i31) invSfericoI(QsfericoRRP,5,10);
```

(%o31)
$$\left[\left[\frac{\pi}{3}, 0, 10\right], \left[\frac{\pi}{3}, \pi, -10\right], \left[\pi, 0, \frac{\pi}{3}\right], \left[0, \pi, -\frac{2\pi}{3}\right]\right]$$

```
(%i32) Qsferico11:Q[sferico](q1,q2,q3,L1,L2)
```

(%o32) $(\cos(q1)\cos(q2), \sin(q1), \cos(q1)\sin(q2), \cos(q1)(\sin(q2)\,q3 + L2\cos(q2));$
$\sin(q1)\cos(q2), -\cos(q1), \sin(q1)\sin(q2), \sin(q1)(\sin(q2)\,q3 + L2\cos(q2)); \sin(q2), 0, -\cos(q2),$
$-\cos(q2)\,q3 + L2\sin(q2) + L1; 0, 0, 0, 1)$

```
(%i33) invSfericoI(Qsferico11,L1,L2);
```

(%o33)
$$\Bigg[\Bigg[\mathrm{atan2}(\sin(q1)(\sin(q2)\,q3 + L2\cos(q2)), \cos(q1)(\sin(q2)\,q3 + L2\cos(q2))),$$
$$\mathrm{atan2}\Bigg(\frac{\sqrt{-\cos(2\,q2)\,q3^2 + q3^2 + 2\,L2\sin(2\,q2)\,q3 + L2^2\cos(2\,q2) + L2^2}\,|q3|}{\sqrt{2}} - L2\cos(q2)\,q3 +$$
$$L2^2\sin(q2), \cos(q2)\,q3\,|q3| - L2\sin(q2)\,|q3| +$$
$$\frac{L2\sqrt{-\cos(2\,q2)\,q3^2 + q3^2 + 2\,L2\sin(2\,q2)\,q3 + L2^2\cos(2\,q2) + L2^2}}{\sqrt{2}}\Bigg), |q3|\Bigg],$$
$$\Bigg[\mathrm{atan2}(\sin(q1)(\sin(q2)\,q3 + L2\cos(q2)), \cos(q1)(\sin(q2)\,q3 + L2\cos(q2))),$$
$$\mathrm{atan2}\Bigg(\frac{\sqrt{-\cos(2\,q2)\,q3^2 + q3^2 + 2\,L2\sin(2\,q2)\,q3 + L2^2\cos(2\,q2) + L2^2}\,|q3|}{\sqrt{2}} - L2\cos(q2)\,q3 +$$
$$L2^2\sin(q2), -\cos(q2)\,q3\,|q3| + L2\sin(q2)\,|q3| -$$
$$\frac{L2\sqrt{-\cos(2\,q2)\,q3^2 + q3^2 + 2\,L2\sin(2\,q2)\,q3 + L2^2\cos(2\,q2) + L2^2}}{\sqrt{2}}\Bigg), -|q3|\Bigg], \Bigg[\mathrm{atan2}\Bigg(0,$$
$$-\frac{\cos(q2)}{\sqrt{1-\sin(q2)^2}}\Bigg), -\mathrm{atan2}\left(\sin(q2), \sqrt{1-\sin(q2)^2}\right), \mathrm{atan2}\Bigg(\frac{\sin(q1)\cos(q2)}{\sqrt{1-\sin(q2)^2}}, \frac{\cos(q1)\cos(q2)}{\sqrt{1-\sin(q2)^2}}\Bigg)\Bigg],$$

$$\left[\operatorname{atan2}\left(0,\frac{\cos(q2)}{\sqrt{1-\sin(q2)^2}}\right),-\operatorname{atan2}\left(\sin(q2),-\sqrt{1-\sin(q2)^2}\right),-\operatorname{atan2}\left(\frac{\sin(q1)\cos(q2)}{\sqrt{1-\sin(q2)^2}},\right.\right.$$
$$\left.\left.-\frac{\cos(q1)\cos(q2)}{\sqrt{1-\sin(q2)^2}}\right)\right]\right]$$

(%i34)

Singolarità

```
Maxima 5.44.0 http://maxima.sourceforge.net
using Lisp SBCL 2.0.0
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
The function bug_report() provides bug reporting information.
```

(%i1) `x:cos (q[1])*(q[3]*sin (q[2])+L[2]*cos (q[2]));`

(%o1) $\cos(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2))$

(%i2) `y:sin (q[1])*(q[3]*sin (q[2])+L[2]*cos (q[2]));`

(%o2) $\sin(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2))$

(%i3) `z:L[2]*sin (q[2])-q[3]*cos (q[2])+L[1];`

(%o3) $L_2\sin(q_2)-q_3\cos(q_2)+L_1$

(%i29)
```
J:matrix([diff(x,q[1]),diff(x,q[2]),diff(x,q[3])],
         [diff(y,q[1]),diff(y,q[2]),diff(y,q[3])],
         [diff(z,q[1]),diff(z,q[2]),diff(z,q[3])]);
```

(%o29) $(-\sin(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2)),\ \cos(q_1)\,(q_3\cos(q_2)-L_2\sin(q_2)),\ \cos(q_1)\sin(q_2);\ \cos(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2)),\ \sin(q_1)\,(q_3\cos(q_2)-L_2\sin(q_2)),\ \sin(q_1)\sin(q_2);\ 0,\ q_3\sin(q_2)+L_2\cos(q_2),\ -\cos(q_2))$

(%i5) `dJ:trigsimp(expand(determinant(J)));`

(%o5) $q_3^2\sin(q_2)+L_2\,q_3\cos(q_2)$

$$q_3^2\sin(q_2)+L_2\,q_3\cos(q_2)\neq 0$$

Supponendo che $q_3\neq 0$:

$$\sin(q_2)=-\frac{L_2\cos(q_2)}{q_3}$$

$$\frac{L_2^2\cos^2(q_2)}{q_3^2}+\cos^2(q_2)=1$$

$$\cos^2(q_2)\left(\frac{q_3^2+L_2^2}{q_3^2}\right)=\frac{q_3^2}{q_3^2+L_2^2}\to\cos(q_2)=\pm\frac{q_3}{\sqrt{q_3^2+L_2^2}}$$

$$\sin(q_2)=\mp\frac{L_2}{\sqrt{q_3^2+L_2^2}}$$

$$q_2=\operatorname{atan2}(\mp L_2,\pm q_3)$$

Le singolarità di velocità di hanno per:

$$q_3=0\wedge q_2=\begin{cases}\operatorname{atan2}(-L_2,q_3)\\ \operatorname{atan2}(+L_2,-q_3)\end{cases}$$

(%i6) `Jq3:subst(q[3]=0,J);`

(%o6) $\begin{pmatrix} -L_2\sin(q_1)\cos(q_2) & -L_2\cos(q_1)\sin(q_2) & \cos(q_1)\sin(q_2) \\ L_2\cos(q_1)\cos(q_2) & -L_2\sin(q_1)\sin(q_2) & \sin(q_1)\sin(q_2) \\ 0 & L_2\cos(q_2) & -\cos(q_2) \end{pmatrix}$

(%i7) `trigsimp(nullspace(Jq3));`

Proviso: $\mathrm{notequal}(\cos(q_1)\sin(q_2),0) \wedge \mathrm{notequal}((L_2\sin(q_1)^2+L_2\cos(q_1)^2)\cos(q_2)\sin(q_2),0)$

(%o7) $\mathrm{span}\left(\begin{pmatrix} 0 \\ L_2\cos(q_2)\sin(q_2) \\ L_2^2\cos(q_2)\sin(q_2) \end{pmatrix}\right)$

Supponendo che $q_2 \neq 0$, le singolarità di velocità di hanno per $v \in \mathrm{Im}\left\{\begin{pmatrix} 0 \\ L_2\cos(q_2)\sin(q_2) \\ L_2^2\cos(q_2)\sin(q_2) \end{pmatrix}\right\}$

Se $q_2 = 0$:

(%i8) `Jq32:subst(q[2]=0,Jq3);`

(%o8) $\begin{pmatrix} -L_2\sin(q_1) & 0 & 0 \\ L_2\cos(q_1) & 0 & 0 \\ 0 & L_2 & -1 \end{pmatrix}$

(%i9) `nullspace(Jq32);`

Proviso: $\mathrm{notequal}(-L_2\sin(q_1),0) \wedge \mathrm{notequal}(-L_2^2\sin(q_1),0)$

(%o9) $\mathrm{span}\left(\begin{pmatrix} 0 \\ -L_2\sin(q_1) \\ -L_2^2\sin(q_1) \end{pmatrix}\right)$

Se $q_3 = 0$, $q_2 = 0$, $q_1 \neq 0$, le singolarità di velocità sono per $v \in \Im m\left\{\begin{pmatrix} 0 \\ -L_2\sin(q_1) \\ -L_2^2\sin(q_1) \end{pmatrix}\right\}$

Se $q_3 = 0$, $q_2 = 0$, $q_1 = 0$:

(%i10) `Jq321:subst(q[1]=0,Jq32);`

(%o10) $\begin{pmatrix} 0 & 0 & 0 \\ L_2 & 0 & 0 \\ 0 & L_2 & -1 \end{pmatrix}$

(%i11) `nullspace(Jq321);`

Proviso: $\mathrm{notequal}(L_2,0) \wedge \mathrm{notequal}(L_2^2,0)$

(%o11) $\mathrm{span}\left(\begin{pmatrix} 0 \\ L_2 \\ L_2^2 \end{pmatrix}\right)$

(%i12)

le singolarità di velocità sono per $v \in \Im m\left\{\begin{pmatrix} 0 \\ L_2 \\ L_2^2 \end{pmatrix}\right\}$

Inoltre se $q_2 = \mathrm{atan2}(-L_2, q_3)$:

(%i14) `J:ratsimp(subst(q[2]=atan2(-L[2]/sqrt(q[3]^2+L[2]^2),q[3]/`
`sqrt(q[3]^2+L[2]^2)),J));`

(%o14) $\begin{pmatrix} 0 & \sqrt{q_3^2+L_2^2}\cos(q_1) & -\frac{L_2\cos(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & \sqrt{q_3^2+L_2^2}\sin(q_1) & -\frac{L_2\sin(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & -\frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$

(%i15) `nullspace(J);`

Proviso: notequal$\left(\sqrt{q_3^2+L_2^2}\cos(q_1), 0\right) \wedge$ notequal$((-q_3^3-L_2^2\,q_3)\cos(q_1), 0)$

(%o15) $\mathrm{span}\left(\begin{pmatrix} (-q_3^3-L_2^2\,q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right)$

Se $q_1 \neq \frac{\pi}{2}$, le singolarità di velocità sono per $v \in \Im m\left\{\begin{pmatrix} (-q_3^3-L_2^2\,q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right\}$.

Se $q_1 = \frac{\pi}{2}$:

(%i20) `Jq1:subst(q[1]=%pi/2,J);`

(%o20) $\begin{pmatrix} 0 & 0 & 0 \\ 0 & \sqrt{q_3^2+L_2^2} & -\frac{L_2}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & -\frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$

(%i21) `nullspace(Jq1);`

Proviso: notequal$\left(\sqrt{q_3^2+L_2^2}, 0\right) \wedge$ notequal$(-q_3^3-L_2^2\,q_3, 0)$

(%o21) $\mathrm{span}\left(\begin{pmatrix} -q_3^3-L_2^2\,q_3 \\ 0 \\ 0 \end{pmatrix}\right)$

le singolarità di velocità sono per $v \in \Im m\left\{\begin{pmatrix} -q_3^3-L_2^2\,q_3 \\ 0 \\ 0 \end{pmatrix}\right\}$

Infine, se $q_2 = \mathrm{atan2}(L_2, -q_3)$:

(%i22) `J:ratsimp(subst(q[2]=atan2(L[2]/sqrt(q[3]^2+L[2]^2),-q[3]/sqrt(q[3]^2+L[2]^2)),J));`

(%o22) $\begin{pmatrix} 0 & \sqrt{q_3^2+L_2^2}\cos(q_1) & -\frac{L_2\cos(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & \sqrt{q_3^2+L_2^2}\sin(q_1) & -\frac{L_2\sin(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & -\frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$

(%i23) `nullspace(J);`

Proviso: notequal$\left(\sqrt{q_3^2+L_2^2}\cos(q_1), 0\right) \wedge$ notequal$((-q_3^3-L_2^2\,q_3)\cos(q_1), 0)$

(%o23) $\mathrm{span}\left(\begin{pmatrix} (-q_3^3-L_2^2\,q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right)$

Se $q_1 \neq \frac{\pi}{2}$, le singolarità di velocità sono per $v \in \Im m\left\{\begin{pmatrix} (-q_3^3-L_2^2\,q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right\}$.

Se $q_1 = \frac{\pi}{2}$:

(%i24) `J:subst(q[1]=%pi/2,J);`

(%o24) $\begin{pmatrix} 0 & 0 & 0 \\ 0 & \sqrt{q_3^2+L_2^2} & -\frac{L_2}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & -\frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$

(%i25) `nullspace(J);`

Proviso: $\mathrm{notequal}\left(\sqrt{q_3^2+L_2^2},0\right)\wedge \mathrm{notequal}(-q_3^3-L_2^2\,q_3,0)$

(%o25) $\mathrm{span}\left(\left(\begin{array}{c}-q_3^3-L_2^2\,q_3\\0\\0\end{array}\right)\right)$

le singolarità di velocità sono per $v\in\Im m\left\{\left(\begin{array}{c}(-q_3^3-L_2^2\,q_3)\cos(q_1)\\0\\0\end{array}\right)\right\}$.

(%i26)

Singolairtà di Forza

```
(%i40) J:-transpose(J);
```

(%o40) $(-\sin(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2)),\cos(q_1)\,(q_3\cos(q_2)-L_2\sin(q_2)),\cos(q_1)\sin(q_2);\cos(q_1)\,(q_3\sin(q_2)+L_2\cos(q_2)),\sin(q_1)\,(q_3\cos(q_2)-L_2\sin(q_2)),\sin(q_1)\sin(q_2);0,q_3\sin(q_2)+L_2\cos(q_2),-\cos(q_2))$

```
(%i31) dJ:trigsimp(expand(determinant(J)));
```

(%o31) $-q_3^2\sin(q_2)-L_2\,q_3\cos(q_2)$

$$-q_3^2\sin(q_2)-L_2\,q_3\cos(q_2)\neq 0$$

Supponendo che $q_3\neq 0$:

$$\sin(q_2)=-\frac{L_2\cos(q_2)}{q_3}$$

$$\frac{L_2^2\cos^2(q_2)}{q_3^2}+\cos^2(q_2)=1$$

$$\cos^2(q_2)\left(\frac{q_3^2+L_2^2}{q_3^2}\right)=\frac{q_3^2}{q_3^2+L_2^2}\rightarrow\cos(q_2)=\pm\frac{q_3}{\sqrt{q_3^2+L_2^2}}$$

$$\sin(q_2)=\mp\frac{L_2}{\sqrt{q_3^2+L_2^2}}$$

$$q_2=\mathrm{atan2}(\mp L_2,\pm q_3)$$

Le singolarità di forza si hanno per:

$$q_3=0\wedge q_2=\begin{cases}\mathrm{atan2}(-L_2,q_3)\\\mathrm{atan2}(+L_2,-q_3)\end{cases}$$

$$-q_3^2\sin(q_2)-L_2\,q_3\cos(q_2)=0$$

Se $q_3\neq 0$:

$$-q_3\sin(q_2)-L_2\cos(q_2)=0$$

$$\sin(q_2)=-\frac{L_2}{q_3}\cos(q_2)$$

$$\sin(q_2)^2+\cos(q_2)^2=1\rightarrow-\frac{L_2^2}{q_3^2}\cos(q_2)^2+\cos(q_2)^2=1$$

```
(%i32) Jq3:subst(q[3]=0,J);
```

(%o32) $\left(\begin{array}{ccc}L_2\sin(q_1)\cos(q_2) & -L_2\cos(q_1)\cos(q_2) & 0\\ L_2\cos(q_1)\sin(q_2) & L_2\sin(q_1)\sin(q_2) & -L_2\cos(q_2)\\ -\cos(q_1)\sin(q_2) & -\sin(q_1)\sin(q_2) & \cos(q_2)\end{array}\right)$

```
(%i33) trigsimp(nullspace(Jq3));
```

Proviso: $\mathrm{notequal}(L_2\sin(q_1)\cos(q_2),0)\wedge\mathrm{notequal}(-L_2^2\sin(q_1)\cos(q_2)^2,0)$

(%o33) $\operatorname{span}\left(\begin{pmatrix} -L_2^2\cos(q_1)\cos(q_2)^2 \\ -L_2^2\sin(q_1)\cos(q_2)^2 \\ -L_2^2\cos(q_2)\sin(q_2) \end{pmatrix}\right)$

(%i34)

$$L_2\sin(q_1)\cos(q_2)\neq 0 \rightarrow q_1\neq 0,\, q_2\neq \frac{\pi}{2}$$

Se $q_1\neq 0, q_2\neq \frac{\pi}{2}$, si hanno le singolarità di forza se $\tau\in \operatorname{Im}\left\{\begin{pmatrix} -L_2^2\cos(q_1)\cos(q_2)^2 \\ -L_2^2\sin(q_1)\cos(q_2)^2 \\ -L_2^2\cos(q_2)\sin(q_2) \end{pmatrix}\right\}$

Se $q_1=0\wedge q_2\neq\frac{\pi}{2}$:

```
(%i34) Jq31:subst(q[1]=0,Jq3);
```

(%o34) $\begin{pmatrix} 0 & -L_2\cos(q_2) & 0 \\ L_2\sin(q_2) & 0 & -L_2\cos(q_2) \\ -\sin(q_2) & 0 & \cos(q_2) \end{pmatrix}$

```
(%i35) nullspace(Jq31);
```

Proviso: notequal$(-L_2\cos(q_2), 0)\wedge$notequal$(L_2^2\cos(q_2)^2, 0)$

(%o35) $\operatorname{span}\left(\begin{pmatrix} L_2^2\cos(q_2)^2 \\ 0 \\ L_2^2\cos(q_2)\sin(q_2) \end{pmatrix}\right)$

(%i36)

Se $q_1=0, q_2\neq\frac{\pi}{2}$, si hanno le singolarità di forza se $\tau\in \operatorname{Im}\left\{\begin{pmatrix} L_2^2\cos(q_2)^2 \\ 0 \\ L_2^2\cos(q_2)\sin(q_2) \end{pmatrix}\right\}$

Se $q_1\neq 0, q_2=\frac{\pi}{2}$:

```
(%i36) Jq32:subst(q[2]=%pi/2,Jq3);
```

(%o36) $\begin{pmatrix} 0 & 0 & 0 \\ L_2\cos(q_1) & L_2\sin(q_1) & 0 \\ -\cos(q_1) & -\sin(q_1) & 0 \end{pmatrix}$

```
(%i37) nullspace(Jq32);
```

Proviso: notequal$(L_2\cos(q_1), 0)$

(%o37) $\operatorname{span}\left(\begin{pmatrix} 0 \\ 0 \\ -L_2\sin(q_1) \end{pmatrix}, \begin{pmatrix} -L_2\sin(q_1) \\ L_2\cos(q_1) \\ 0 \end{pmatrix}\right)$

si hanno le singolarità di forza se $\tau\in \operatorname{Im}\left\{\begin{pmatrix} 0 \\ 0 \\ -L_2\sin(q_1) \end{pmatrix}, \begin{pmatrix} -L_2\sin(q_1) \\ L_2\cos(q_1) \\ 0 \end{pmatrix}\right\}$

Se $q_1=0, q_2=\frac{\pi}{2}$:

```
(%i38) Jq321:subst([q[2]=%pi/2,q[1]=0],Jq3);
```

(%o38) $\begin{pmatrix} 0 & 0 & 0 \\ L_2 & 0 & 0 \\ -1 & 0 & 0 \end{pmatrix}$

```
(%i39) nullspace(Jq321);
```

Proviso: notequal$(L_2, 0)$

(%o39) $\mathrm{span}\left(\begin{pmatrix} 0 \\ 0 \\ L_2 \end{pmatrix}, \begin{pmatrix} 0 \\ L_2 \\ 0 \end{pmatrix}\right)$

si hanno le singolarità di forza se $\tau \in \mathrm{Im}\left\{\begin{pmatrix} 0 \\ 0 \\ L_2 \end{pmatrix}, \begin{pmatrix} 0 \\ L_2 \\ 0 \end{pmatrix}\right\}$

Se $q_2 = \mathrm{atan2}(-L_2, q_3)$:

```
(%i43) Jq2:subst(q[2]=atan2(-L[2],q[3]),J);
```

(%o43)
$$\begin{pmatrix} 0 & \left(\frac{q_3^2}{\sqrt{q_3^2+L_2^2}} + \frac{L_2^2}{\sqrt{q_3^2+L_2^2}}\right)\cos(q_1) & -\frac{L_2\cos(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & \left(\frac{q_3^2}{\sqrt{q_3^2+L_2^2}} + \frac{L_2^2}{\sqrt{q_3^2+L_2^2}}\right)\sin(q_1) & -\frac{L_2\sin(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & -\frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$$

```
(%i44) nullspace(Jq2)
```

Proviso: $\mathrm{notequal}\left(\sqrt{q_3^2+L_2^2}\cos(q_1), 0\right) \wedge \mathrm{notequal}((-q_3^3 - L_2^2 q_3)\cos(q_1), 0)$

(%o44) $\mathrm{span}\left(\begin{pmatrix} (-q_3^3 - L_2^2 q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right)$

Se $q_1 \neq 0$ si hanno le singolarità di forza se $\tau \in \mathrm{Im}\left\{\begin{pmatrix} (-q_3^3 - L_2^2 q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right\}$.

Se $q_1 = 0$:

```
(%i45) Jq21:subst(q[1]=0,Jq2);
```

(%o45)
$$\begin{pmatrix} 0 & \frac{q_3^2}{\sqrt{q_3^2+L_2^2}} + \frac{L_2^2}{\sqrt{q_3^2+L_2^2}} & -\frac{L_2}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & 0 \\ 0 & 0 & -\frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$$

```
(%i46) nullspace(Jq21);
```

Proviso: $\mathrm{notequal}\left(\sqrt{q_3^2+L_2^2}, 0\right) \wedge \mathrm{notequal}(-q_3^3 - L_2^2 q_3, 0)$

(%o46) $\mathrm{span}\left(\begin{pmatrix} -q_3^3 - L_2^2 q_3 \\ 0 \\ 0 \end{pmatrix}\right)$

si hanno le singolarità di forza se $\tau \in \mathrm{Im}\left\{\begin{pmatrix} -q_3^3 - L_2^2 q_3 \\ 0 \\ 0 \end{pmatrix}\right\}$

Se $q_2 = \mathrm{atan2}(L_2, -q_3)$:

```
(%i47) Jq2:subst(q[2]=atan2(L[2],-q[3]),J);
```

(%o47)
$$\begin{pmatrix} 0 & \left(-\frac{q_3^2}{\sqrt{q_3^2+L_2^2}} - \frac{L_2^2}{\sqrt{q_3^2+L_2^2}}\right)\cos(q_1) & \frac{L_2\cos(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & \left(-\frac{q_3^2}{\sqrt{q_3^2+L_2^2}} - \frac{L_2^2}{\sqrt{q_3^2+L_2^2}}\right)\sin(q_1) & \frac{L_2\sin(q_1)}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & \frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$$

```
(%i48) nullspace(Jq2);
```

Proviso: $\mathrm{notequal}\left(-\sqrt{q_3^2+L_2^2}\cos(q_1), 0\right) \wedge \mathrm{notequal}((-q_3^3 - L_2^2 q_3)\cos(q_1), 0)$

(%o48) $\mathrm{span}\left(\begin{pmatrix} (-q_3^3 - L_2^2\, q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right)$

Se $q_1 \neq 0$ si hanno le singolarità di forza se $\tau \in \mathrm{Im}\left\{\begin{pmatrix} (-q_3^3 - L_2^2\, q_3)\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right\}$.

Se $q_1 = 0$:

```
(%i49) Jq21:subst(q[1]=0,Jq2);
```

(%o49) $\begin{pmatrix} 0 & -\frac{q_3^2}{\sqrt{q_3^2+L_2^2}} - \frac{L_2^2}{\sqrt{q_3^2+L_2^2}} & \frac{L_2}{\sqrt{q_3^2+L_2^2}} \\ 0 & 0 & 0 \\ 0 & 0 & \frac{q_3}{\sqrt{q_3^2+L_2^2}} \end{pmatrix}$

```
(%i50) nullspace(Jq21);
```

Proviso: $\mathrm{notequal}\left(-\sqrt{q_3^2+L_2^2}, 0\right) \wedge \mathrm{notequal}(-q_3^3 - L_2^2\, q_3, 0)$

(%o50) $\mathrm{span}\left(\begin{pmatrix} -q_3^3 - L_2^2\, q_3 \\ 0 \\ 0 \end{pmatrix}\right)$

si hanno le singolarità di forza se $\tau \in \mathrm{Im}\left\{\begin{pmatrix} -q_3^3 - L_2^2\, q_3 \\ 0 \\ 0 \end{pmatrix}\right\}$.

```
(%i51)
```