Cinematica diretta SCARA

***N.B.:* le grandezze diverse da quelle di giunto q_i sono L_i, D_i. Esse sono rispettivamente la distanza tra i sistemi di riferimento R_i e R_{i+1} nelle operazioni della matrice avvitamento $A_z(\theta, d)$ e $A_x(\alpha, a)$.**



	ϑ	d	α	a
1	q_1	L_1	0	D_1
2	q_2	0	0	D_2
3	0	q_3	0	0

Tabella 1.

Funzioni ausiliarie:

```
(%i1) inverseLaplace(SI,theta):=block([res],
                                 M:SI,
                                 MC:SI,
                                 for i:1 thru 3 do(
                                   for j:1 thru 3 do
                                      (
                                        aC:M[i,j],
                                        b:ilt(aC,s,theta),
                                        MC[i,j]:b
                                       )
                                   ),
                                 res:MC

                              )
```

(%o1) inverseLaplace(SI, ϑ) := **block** ([res], M: SI, MC: SI, **for** i **thru** 3 **do for** j **thru** 3 **do** (aC: $M_{i,j}$, b: ilt(aC, s, ϑ), $\mathrm{MC}_{i,j}$: b), res: MC)

(%i2)
```
rotLaplace(k,theta):=block([res],
                          S:ident(3),
                          I:ident(3),
                        for i:1 thru 3 do
                          (
                          for j:1 thru 3 do
                              (
                                if i=j
                                   then S[i][j]:0
                                elseif j>i
                                   then (
                                  temp:(-1)^(j-i)*k[3-remainder(i+j,3)],
                                         S[i][j]:temp,
                                         S[j][i]:-temp

                                          )
                                )
                           ),
                          res:inverseLaplace(invert(s*I-S),theta)


                        )
```

(%o2) rotLaplace(k,ϑ) := **block** ([res], S: ident(3), I: ident(3), **for** i **thru** 3 **do for** j **thru** 3 **do if** $i=j$ **then** $(S_i)_j$: 0 **elseif** $j>i$ **then** (temp: $(-1)^{j-i}\,k_{3-\mathrm{remainder}(i+j,3)}$, $(S_i)_j$: temp, $(S_j)_i$: $-$temp), res: inverseLaplace(invert($s\,I-S$), ϑ))

(%i3)
```
Av(v,theta,d):=block([res],
                          Trot:rotLaplace(v,theta),
                          row:matrix([0,0,0,1]),
                          Atemp:addcol(Trot,d*transpose(v)),
                          A:addrow(Atemp,row),
                          res:A
                          )
```

(%o3) Av(v,ϑ,d) := **block** ([res], Trot: rotLaplace(v,ϑ), row: $\begin{pmatrix} 0 & 0 & 0 & 1 \end{pmatrix}$, Atemp: addcol(Trot, d transpose(v)), A: addrow(Atemp, row), res: A)

(%i4)
```
Q(theta,d,alpha,a):=block([res],
                               tempMat:Av([0,0,1],theta,d).Av([1,0,0],alpha,a),
                               Qtrasf:zeromatrix(4,4),
                               for i:1 thru 4 do
                          (
                          for j:1 thru 4 do
                              (
                                Qtrasf[i][j]:trigreduce(tempMat[i][j])
                                          )
                              ),

                                res:Qtrasf
                                )
```

(%o4) $Q(\vartheta,d,\alpha,a)$:= **block** ([res], tempMat: Av$([0,0,1],\vartheta,d)\cdot$Av$([1,0,0],\alpha,a)$, Qtrasf: zeromatrix(4, 4), **for** i **thru** 4 **do for** j **thru** 4 **do** $(\mathrm{Qtrasf}_i)_j$: trigreduce($(\mathrm{tempMat}_i)_j$), res: Qtrasf)

(%i5)
```
let(sin(q[1]), s[1]);
```

(%o5) $\sin(q_1) \longrightarrow s_1$

(%i6) `let(sin(q[2]), s[2]);`

(%o6) $\sin(q_2) \longrightarrow s_2$

(%i7) `let(cos(q[1]),c[1]);`

(%o7) $\cos(q_1) \longrightarrow c_1$

(%i8) `let(cos(q[2]),c[2]);`

(%o8) $\cos(q_2) \longrightarrow c_2$

(%i9) `let(sin(q[1]+q[2]), s[12]);`

(%o9) $\sin(q_2+q_1) \longrightarrow s_{12}$

(%i10) `let(cos(q[1]+q[2]),c[12]);`

(%o10) $\cos(q_2+q_1) \longrightarrow c_{12}$

(%i11)

(%i11)

Cinematica diretta:

```
(%i11) Q[scara](q1,q2,q3,L1,D1,D2):=trigsimp(trigrat(trigreduce(trigexpand(
                        Q(q1,L1,0,D1).
                         Q(q2,0,0,D2).
                         Q(0,q3,0,0)
                         ))));
```

(%o11) $Q_{\mathrm{scara}}(q1,q2,q3,L1,D1,D2) := \mathrm{trigsimp}(\mathrm{trigrat}(\mathrm{trigreduce}(\mathrm{trigexpand}(Q(q1,L1,0,D1)\cdot Q(q2,0,0,D2)\cdot Q(0,q3,0,0)))))$

(%i12) `Qscara:Q[scara](q[1],q[2],q[3],L[1],D[1],D[2]);`

(%o12) $$\begin{pmatrix} \cos(q_2+q_1) & -\sin(q_2+q_1) & 0 & D_2\cos(q_2+q_1)+D_1\cos(q_1) \\ \sin(q_2+q_1) & \cos(q_2+q_1) & 0 & D_2\sin(q_2+q_1)+D_1\sin(q_1) \\ 0 & 0 & 1 & q_3+L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

(%i13) `letsimp(Qscara);`

(%o13) $$\begin{pmatrix} c_{12} & -s_{12} & 0 & D_2\,c_{12}+D_1\,c_1 \\ s_{12} & c_{12} & 0 & D_2\,s_{12}+D_1\,s_1 \\ 0 & 0 & 1 & q_3+L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

(%i14)

Cinematica inversa SCARA

Al fine di risolvere il problema di cinematica inversa del robot scara occorre risolevere il problema di posizione ed orientamente inverso.Inizialmente individuare lo spazio di lavoro, le soluzioni generiche, singolari ed infine le variabili di giunto q_i ed in seguito determinare l'orientamento del robot.

Dalla cinematica diretta del robot SCARA sappiamo che:

$$Q_{\mathrm{SCARA}} = \begin{pmatrix} c_{12} & -s_{12} & 0 & D_2\,c_{12}+D_1\,c_1 \\ s_{12} & c_{12} & 0 & D_2\,s_{12}+D_1\,s_1 \\ 0 & 0 & 1 & q_3+L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Cinematica inversa di posizione

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} D_2\, c_{12} + D_1\, c_1 \\ D_2\, s_{12} + D_1\, s_1 \\ q_3 + L_1 \end{pmatrix}$$

Dato chew L_1 è nota è possibile ottenere la varibiale di giungo q_3:

$$q_3 = z$$

e, riscrivendo le equazioni rimanenti, si ottiene il seguente sistema:

$$\begin{cases} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} D_2 \cos(q_2 + q_1) + D_1 \cos(q_1) \\ D_2 \sin(q_2 + q_1) + D_1 \sin(q_1) \\ 0 \end{pmatrix} \\ q_3 = z \end{cases}$$

Occorre ora calcolare le variabili di giunto q_1, q_2. Il problema è equivalente al problema di cinematica di posizione inversa del 2DOF planare. Infatti:

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} D_2 \cos(q_2 + q_1) + D_1 \cos(q_1) \\ D_2 \sin(q_2 + q_1) + D_1 \sin(q_1) \\ 0 \end{pmatrix}$$

$$\begin{pmatrix} x \\ y \end{pmatrix} = R(q_1 + q_2) \begin{pmatrix} D_2 \\ 0 \end{pmatrix} + R(q_1) \begin{pmatrix} D_1 \\ 0 \end{pmatrix}$$

Poiché $R(q_1 + q_2) = R(q_1)\, R(q_2)$, è possibile mettere in evidenza $R(q_1)$:

$$\begin{pmatrix} x \\ y \end{pmatrix} = R(q_1) \left\{ R(q_2) \begin{pmatrix} D_2 \\ 0 \end{pmatrix} + \begin{pmatrix} D_1 \\ 0 \end{pmatrix} \right\}$$

La matrice di rotazione $R(q_1)$ ha non varia la norma del vettore $\begin{pmatrix} \mathrm{x} \\ \mathrm{y} \end{pmatrix}$ e $R(q_2) \begin{pmatrix} D_2 \\ 0 \end{pmatrix} + \begin{pmatrix} D_1 \\ 0 \end{pmatrix}$. Quindi possiamo imporre che abbiano la stessa norma. In particolare:

$$\left|\left| \begin{pmatrix} x \\ y \end{pmatrix} \right|\right|_2 = \left|\left| R(q_2) \begin{pmatrix} D_2 \\ 0 \end{pmatrix} + \begin{pmatrix} D_1 \\ 0 \end{pmatrix} \right|\right|_2$$

$$\begin{pmatrix} D_2 & 0 \end{pmatrix} R(q_2)^T R(q_2) \begin{pmatrix} D_2 \\ 0 \end{pmatrix} + \begin{pmatrix} D_1 & 0 \end{pmatrix} \begin{pmatrix} D_1 \\ 0 \end{pmatrix} + 2 \begin{pmatrix} D_1 & 0 \end{pmatrix} R(q_2) \begin{pmatrix} D_2 \\ 0 \end{pmatrix}$$

$$x^2 + y^2 = D_2^2 + D_1^2 + 2\, D_1\, D_2 \cos(q_2)$$

$$\cos(q_2) = \frac{x^2 + y^2 - D_2^2 - D_1^2}{2\, D_1\, D_2}$$

Poiché $-1 \leqslant \cos(q_2) \leqslant 1$, otteniamo infine lo spazio operativo del robot SCARA:

$$-1 \leqslant \frac{x^2 + y^2 - D_2^2 - D_1^2}{2\, D_1\, D_2} \leqslant 1$$

$$D_2^2 + D_1^2 - 2\, D_1\, D_2 \leqslant x^2 + y^2 \leqslant 2\, D_1\, D_2 + D_2^2 + D_1^2$$

Le disequazioni delimitano due circonferenze di raggio $D_2 + D_1$ e $D_2 - D_1$, in cui la regione valida presa inconsiderazione è quella regione di spazio compresa tra le curve. Infine, considerando, a differenza del robot 2DOF planare, anche la variabile q_3 si delimita un cilindro cavo di altezza q_3 e di raggio interno $D_2 - D_1$ e di raggio esterno $D_2 + D_1$.

A questo punto è possibile determinare l'espressione della variabile di giunto q_2:

$$\cos(q_2) = \frac{x^2 + y^2 - D_2^2 - D_1^2}{2\, D_1\, D_2}$$

$$\sin(q_2) = \pm\sqrt{1 - \cos(q_2)^2} = \pm\sqrt{1 - \frac{x^2 + y^2 - D_2^2 - D_1^2}{2\, D_1\, D_2}}$$

In particolare, si hanno due soluzioni generiche se $\left|\frac{x^2 + y^2 - D_2^2 - D_1^2}{2\, D_1\, D_2}\right| \neq 1$. Nel caso in cui $\left|\frac{x^2 + y^2 - D_2^2 - D_1^2}{2\, D_1\, D_2}\right| = 1$, si ha una soluzione singolare.

A questo punto la quantità $R(q_2)\begin{pmatrix} D_2 \\ 0 \end{pmatrix} + \begin{pmatrix} D_1 \\ 0 \end{pmatrix}$ è nota:

$$\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} \cos(q_1) & -\sin(q_1) \\ \sin(q_1) & \cos(q_1) \end{pmatrix} \begin{pmatrix} D_2 \cos(q_2) + D_1 \\ \sin(q_2)\, D_2 \end{pmatrix}$$

$$\begin{pmatrix} \cos(q_1) \\ \sin(q_1) \end{pmatrix} = \frac{1}{(D_2 \cos(q_2) + D_1)^2 + D_2^2 \sin(q_2)^2} \begin{pmatrix} D_2 \cos(q_2) + D_1 & \sin(q_2)\, D_2 \\ -\sin(q_2)\, D_2 & D_2 \cos(q_2) + D_1 \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix}$$

$$= \frac{1}{D_2^2 \cos(q_2)^2 + D_1^2 + 2\, D_2 D_1 \cos(q_2) + D_1 + D_2^2 \sin(q_2)^2} \begin{pmatrix} (D_2 \cos(q_2) + D_1)\, x + \sin(q_2)\, D_2\, y \\ -\sin(q_2)\, D_2\, x + (D_2 \cos(q_2) + D_1)\, y \end{pmatrix}$$

Da cui si ottiene la variabile di giunto q_1;

$$q_1 = \operatorname{atan2}\!\left(\frac{-\sin(q_2)\, D_2\, x + (D_2 \cos(q_2) + D_1)\, y}{D_2^2 \cos(q_2)^2 + D_1^2 + D_2^2 \sin(q_2)^2}, \frac{(D_2 \cos(q_2) + D_1)\, x + \sin(q_2)\, D_2\, y}{D_2^2 \cos(q_2)^2 + D_1^2 + D_2^2 \sin(q_2)^2} \right)$$

Poiché la quantità $D_2^2 \cos(q_2)^2 + D_1^2 + D_2^2 \sin(q_2)^2 > 0$ è possibile semplificara all'interno della funzione atan2:

$$q_1 = \operatorname{atan2}(\sin(q_2)\, D_2\, x - (D_2 \cos(q_2) + D_1)\, y, (D_2 \cos(q_2) + D_1)\, x - \sin(q_2)\, D_2\, y)$$

Il problema della cinematica inversa ed, in particolare il problema di posizione inverso, per il robot SCARA è risolto.

Orientamento inverso

La risoluzione del problema di orientamento inverso si basa sulla scelta di una terna di Eulero o nautica in condizione non singolari, se possibile.

$$R_{\text{SCARA}} = \begin{pmatrix} \cos(q_2+q_1) & -\sin(q_2+q_1) & 0 \\ \sin(q_2+q_1) & \cos(q_2+q_1) & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$$R_{\text{zyx}} = \begin{pmatrix} c_y c_z & \cdots\cdots\cdots & \\ c_y s_z & \cdots\cdots\cdots & \\ -s_y & s_x c_y & c_x c_y \end{pmatrix}$$

Poiché l'elemento $-s_y = 0 \neq \pm 1$ è possibile risolvere il problema di orientamento inverso con la terna nautica zyx. In particolare:

$$s_y = 0 \longrightarrow c_y = \pm 1 \longrightarrow \phi_y = \text{atan2}(s_y, c_y) \qquad \longrightarrow \quad \phi_y = \begin{cases} 0 \\ \pi \end{cases}$$

$$\begin{cases} s_x c_y = 0 \\ c_x c_y = 1 \end{cases} \longrightarrow \begin{cases} s_x = 0 \\ c_x = \pm 1 \end{cases} \longrightarrow \phi_x = \text{atan2}(s_x, \pm c_x) \quad \longrightarrow \quad \phi_x = \begin{cases} 0 \\ \pi \end{cases}$$

$$\begin{cases} c_y c_z = \cos(q_2+q_1) \\ c_y s_z = \sin(q_2+q_1) \end{cases} \longrightarrow \begin{cases} c_z = \pm\cos(q_1+q_2) \\ s_z = \pm\sin(q_1+q_2) \end{cases} \longrightarrow \phi_z = \text{atan2}(\pm\sin(q_1+q_2), \pm\cos(q_1+q_2))$$

$$\phi_z = \begin{cases} q_1 + q_2 \\ q_1 + q_2 + \pi \end{cases}$$

Riassumendo:

$$\begin{pmatrix} 0 \\ 0 \\ q_1 + q_2 \end{pmatrix}, \begin{pmatrix} \pi \\ \pi \\ q_1 + q_2 + \pi \end{pmatrix}$$

In alternativa, tramite una scelta di una terna di Eulero:

$$R_{\text{xzx}} = \begin{pmatrix} \cos(\beta) & -\sin(\beta)\cos(\gamma) & \sin(\beta)\sin(\gamma) \\ \cos(\alpha)\sin(\beta) & \cdots & \cdots \\ \sin(\alpha)\sin(\beta) & \cdots & \cdots \end{pmatrix}$$

$$R_{\text{zyx}} = \begin{pmatrix} c_y c_z & \cdots\cdots\cdots & \\ c_y s_z & \cdots\cdots\cdots & \\ -s_y & s_x c_y & c_x c_y \end{pmatrix}$$

$$\cos(\beta) = \cos(q_1+q_2) \quad \rightarrow \quad \sin(\beta) = \pm\sqrt{1 - \cos(q_1+q_2)^2} = \pm\sin(q_1+q_2)$$

La configurazione scelta è singolare se:

$$1 - \cos(q_1+q_2)^2 = 0 \rightarrow \cos(q_1+q_2) = \pm 1 \rightarrow q_1 + q_2 = \begin{cases} 0 \\ \pi \end{cases}$$

Supponiamo quindi che $q_1+q_2 \neq \left\{ \begin{smallmatrix} 0 \\ \pi \end{smallmatrix} \right.$:

$$\beta = \mathrm{atan2}\Big(\sqrt{1-\cos(q_1+q_2)^2}, \cos(q_1+q_2)\Big)$$

$$\beta = \left\{ \begin{array}{l} q_1+q_2 \\ q_1+q_2+\pi \end{array} \right.$$

$$\left\{ \begin{array}{l} \cos(\alpha)\sin(\beta) = \sin(q_2+q_1) \\ \sin(\alpha)\sin(\beta) = 0 \end{array} \right. \rightarrow \left\{ \begin{array}{l} \cos(\alpha) = \pm 1 \\ \sin(\alpha) = 0 \end{array} \right.$$

$$\alpha = \mathrm{atan2}(0, \pm 1) = \left\{ \begin{array}{l} 0 \\ \pi \end{array} \right.$$

$$\left\{ \begin{array}{l} -\sin(\beta)\cos(\gamma) = -\sin(q_2+q_1) \\ \sin(\beta)\sin(\gamma) = 0 \end{array} \right. \rightarrow \left\{ \begin{array}{l} \cos(\gamma) = \pm 1 \\ \sin(\gamma) = 0 \end{array} \right.$$

$$\gamma = \mathrm{atan2}(0, \pm 1) = \left\{ \begin{array}{l} 0 \\ \pi \end{array} \right.$$

Riassumendo, sotto l'ipotesi che $q_1+q_2 \neq \left\{ \begin{smallmatrix} 0 \\ \pi \end{smallmatrix} \right.$:

$$\begin{pmatrix} 0 \\ 0 \\ q_1+q_2 \end{pmatrix}, \begin{pmatrix} \pi \\ \pi \\ q_1+q_2+\pi \end{pmatrix}$$

```
(%i14) isRotation(M):=block([MC,res],
                        I:ident(3),
                        MC:ident(3),
                        for i:1 thru 3 do
                        (
                        for j:1 thru 3 do
                           (
                              MC[i][j]:M[i][j]
                              )
                           ),
                        MMT:trigsimp(expand(MC.transpose(MC))),
                        detM:trigsimp(expand(determinant(MC))),

                        if MMT=I  and detM=1
                           then(

                                return(res:1)
                                )

                        else(

                              res: "R is not rotation matrix"
                              )
                        )
```

(%o14) $\mathrm{isRotation}(M) := \textbf{block}\ ([\mathrm{MC}, \mathrm{res}], I\colon \mathrm{ident}(3), \mathrm{MC}\colon \mathrm{ident}(3),$

for i **thru** 3 **do for** j **thru** 3 **do** $(\mathrm{MC}_i)_j$: $(M_i)_j$, MMT: trigsimp(expand(MC · transpose(MC))), detM: trigsimp(expand(determinant(MC))), **if** MMT = $I \wedge$ detM = 1 **then** return(res: 1) **else** res: R is not rotation matrix)

(%i80)
```
calculate(x,y,L1,L2,z):=block([q2plus,q2minus,q1,res],
                              c2:(x^(2)+y^(2)-L1^(2)-L2^(2))/(2*L1*L2),
                              s2:sqrt(1-c2^2),

                              c1Num:combine(expand((L2*c2+L1)*x+s2*L2*y)),
                              s1Num:combine(expand(-s2*L2*x+(L2*c2+L1)*y)),

q1Den:L2^(2)*c2^(2)+L1^(2)+L2^(2)*s2^(2)+2*L1*L2*c2,
                       if abs(c2)=1 then print("La soluzione è singolare")
                              elseif q1Den>0 then(
                                  print("La soluzione non è singolare"),
                                  q1:atan2(ratsimp(s1Num),ratsimp(c1Num)),
                                  q2alto:atan2(ratsimp(s2),ratsimp(c2)),
                                  q2basso:atan2(ratsimp(-s2),ratsimp(c2)),
                                  res:[[q2alto,q1,z],[q2basso,q1,z]])
                              else (
                                  q1:atan2(s1Num/q1Den,c1Num/q1Den),
                                  q2alto:atan2(ratsimp(s2),ratsimp(c2)),
                                  q2basso:atan2(ratsimp(-s2),ratsimp(c2)),
                                  res:[[q2alto,q1],[q2basso,q1]]
                                  )



                              )
```

(%o80) calculate$(x, y, L1, L2, z)$:= **block** $\Big($ [q2plus, q2minus, $q1$, res], $c2$: $\dfrac{x^2+y^2-L1^2-L2^2}{2\,L1\,L2}$, $s2$: $\sqrt{1-c2^2}$, c1Num: combine(expand($(L2\,c2+L1)\,x+s2\,L2\,y$)), s1Num: combine(expand($(-s2)\ L2\ x+(L2\ c2+L1)\ y$)), q1Den: $L2^2\ c2^2+L1^2+L2^2\ s2^2+2\ L1\ L2\ c2$, **if** $|c2|=1$ **then** print(La soluzione è singolare) **elseif** q1Den >0 **then** (print(La soluzione non è singolare), $q1$: atan2(ratsimp(s1Num), ratsimp(c1Num)), q2alto: atan2(ratsimp($s2$), ratsimp($c2$)), q2basso: atan2(ratsimp($-s2$), ratsimp($c2$)), res: [[q2alto, $q1$, z], [q2basso, $q1$, z]]) **else** $\Big($ $q1$: atan2$\left(\dfrac{s\text{1Num}}{q\text{1Den}}, \dfrac{c\text{1Num}}{q\text{1Den}}\right)$, q2alto: atan2(ratsimp($s2$), ratsimp($c2$)), q2basso: atan2(ratsimp($-s2$), ratsimp($c2$)), res: [[q2alto, $q1$], [q2basso, $q1$]] $\Big)\Big)$

```
(%i96) orientation(Qdiretta):=block([sx,cx,sy,cy,phiy1,phiy2,phiz1,phiz2,phix1,
       phix2,sz,sxfirst,second,res],


                                   sy:Qdiretta[3][1],
                                   cy:sqrt(1-sy^2),
                                   phiy1:atan2(sy,cy),
                                   phiy2:atan2(sy,-cy),
                                   sx:Qdiretta[3][2],
                                   cx:Qdiretta[3][3],
                                   phix1:atan2(sx,cx),
                                   phix2:atan2(sx,-cx),
                                   cz:Qdiretta[1][1],
                                   sz:Qdiretta[2][1],
                                   phiz1:atan2(sz,cz),
                                   phiz2:atan2(-sz,-cz),
                                   first:[phix1,phiy1,phiz1],
                                   second:[phix2,phiy2,phiz2],

                                   res:[first,second]

       );
```

(%o96) orientation(Qdiretta) := **block** ([sx, cx, sy, cy, phiy1, phiy2, phiz1, phiz2, phix1, phix2, sz, sxfirst, second, res], sy: $(\mathrm{Qdiretta}_3)_1$, cy: $\sqrt{1-\mathrm{sy}^2}$, phiy1: atan2(sy, cy), phiy2: atan2(sy, −cy), sx: $(\mathrm{Qdiretta}_3)_2$, cx: $(\mathrm{Qdiretta}_3)_3$, phix1: atan2(sx, cx), phix2: atan2(sx, −cx), cz: $(\mathrm{Qdiretta}_1)_1$, sz: $(\mathrm{Qdiretta}_2)_1$, phiz1: atan2(sz, cz), phiz2: atan2(−sz, −cz), first: [phix1, phiy1, phiz1], second: [phix2, phiy2, phiz2], res: [first, second])

```
(%i97) invScara(Qdiretta,D1,D2):=block([pos1,pos2,orien1,orien2,res],
                                  rotation:isRotation(Qdiretta),
                                  if rotation=1 then(
                                     x:Qdiretta[1][4],
                                     y:Qdiretta[2][4],
                                     z:Qdiretta[3][4],
                                     circleInt:D1^(2)+D2^(2)-2*D1*D2,
                                     circleEst:D1^(2)+D2^(2)+2*D1*D2,
                                   print("Il punto x,y è nello spazio di lavoro"),
                                   pos:calculate(x,y,D1,D2,z),
                                   pos1:pos[1],
                                   pos2:pos[2],
                                   orien:orientation(Qdiretta),
                                   orien1:orien[1],
                                   orien2:orien[2],
                                   res:[pos1,pos2,orien1,orien2]

                                  )
                                  else res:rotation
                                  );
```

(%o97) invScara(Qdiretta, $D1$, $D2$) := **block** ([pos1, pos2, orien1, orien2, res], rotation: isRotation(Qdiretta), **if** rotation = 1 **then** (x: $(\mathrm{Qdiretta}_1)_4$, y: $(\mathrm{Qdiretta}_2)_4$, z: $(\mathrm{Qdiretta}_3)_4$, circleInt: $D1^2 + D2^2 + (-2)\ D1\ D2$, circleEst: $D1^2 + D2^2 + 2\ D1\ D2$, print(Il punto x,y è nello spazio di lavoro), pos: calculate($x, y, D1, D2, z$), pos1: pos_1, pos2: pos_2, orien: orientation(Qdiretta), orien1: orien_1, orien2: orien_2, res: [pos1, pos2, orien1, orien2]) **else** res: rotation)

```
(%i98) Qdiretta:Q[scara](%pi/3,0,10,5,15,5)
```

$$\text{(\%o98)}\quad \begin{pmatrix} \frac{1}{2} & -\frac{\sqrt{3}}{2} & 0 & 10 \\ \frac{\sqrt{3}}{2} & \frac{1}{2} & 0 & \frac{5\,3^{\frac{3}{2}}}{2}+\frac{5\sqrt{3}}{2} \\ 0 & 0 & 1 & 15 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(%i99) invScara(Qdiretta,15,5);
```

Il punto x,y è nello spazio di lavoro

La soluzione non è singolare

$$\text{(\%o99)}\quad \left[\left[0,\frac{\pi}{3},15\right],\left[0,\frac{\pi}{3},15\right],\left[0,0,\frac{\pi}{3}\right],\left[\pi,\pi,-\frac{2\pi}{3}\right]\right]$$

```
(%i100)
```

Singolarià

```
(%i1) x:D[2]*cos (q[2]+q[1])+D[1]*cos (q[1]);
```

$$\text{(\%o1)}\quad D_2\cos(q_2+q_1)+D_1\cos(q_1)$$

```
(%i2) y:D[2]*sin (q[2]+q[1])+D[1]*sin (q[1]);
```

$$\text{(\%o2)}\quad D_2\sin(q_2+q_1)+D_1\sin(q_1)$$

```
(%i3) z:q[3]+L[1];
```

$$\text{(\%o3)}\quad q_3+L_1$$

```
(%i4) J:matrix([diff(x,q[1]),diff(x,q[2]),diff(x,q[3])],
               [diff(y,q[1]),diff(y,q[2]),diff(y,q[3])],
               [diff(z,q[1]),diff(z,q[2]),diff(z,q[3])]);
```

$$\text{(\%o4)}\quad \begin{pmatrix} -D_2\sin(q_2+q_1)-D_1\sin(q_1) & -D_2\sin(q_2+q_1) & 0 \\ D_2\cos(q_2+q_1)+D_1\cos(q_1) & D_2\cos(q_2+q_1) & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i5) dJ:trigreduce(expand(determinant(J)));
```

$$\text{(\%o5)}\quad D_1 D_2\sin(q_2)$$

Singolarità se $\det(J)=0 \rightarrow \sin(q_2)=0 \longrightarrow q_2=\begin{cases} 0 \\ \pi \end{cases}$

```
(%i6) Jq:trigreduce(subst(q[2]=0,J));
```

$$\text{(\%o6)}\quad \begin{pmatrix} -D_2\sin(q_1)-D_1\sin(q_1) & -D_2\sin(q_1) & 0 \\ D_2\cos(q_1)+D_1\cos(q_1) & D_2\cos(q_1) & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
(%i7) nullspace(Jq);
```

Proviso: $\mathrm{notequal}(-D_2\sin(q_1),0)\wedge\mathrm{notequal}(-D_2\sin(q_1),0)$

$$\text{(\%o7)}\quad \mathrm{span}\left(\begin{pmatrix} -D_2\sin(q_1) \\ (D_2+D_1)\sin(q_1) \\ 0 \end{pmatrix}\right)$$

Se $q_1\neq 0$, le singolarità di velocità si hanno per $v\in \mathrm{Im}\left\{\begin{pmatrix} -D_2\sin(q_1) \\ (D_2+D_1)\sin(q_1) \\ 0 \end{pmatrix}\right\}$.

Se $q_1=0$:

```
(%i8) Jq1:subst(q[1]=0,Jq);
```

(%o8) $\begin{pmatrix} 0 & 0 & 0 \\ D_2+D_1 & D_2 & 0 \\ 0 & 0 & 1 \end{pmatrix}$

```
(%i9) nullspace(Jq1);
```

Proviso: notequal$(D_2,0) \wedge$ notequal$(D_2,0)$

(%o9) $\operatorname{span}\left(\begin{pmatrix} D_2 \\ -D_2-D_1 \\ 0 \end{pmatrix}\right)$

Se $q_1 \neq 0$ le singolarità di velocità per $v \in \Im m\left\{\begin{pmatrix} D_2 \\ -D_2-D_1 \\ 0 \end{pmatrix}\right\}$

```
(%i10) Jq1:subst(q[1]=0,Jq);
```

(%o10) $\begin{pmatrix} 0 & 0 & 0 \\ D_2+D_1 & D_2 & 0 \\ 0 & 0 & 1 \end{pmatrix}$

```
(%i11) nullspace(Jq1);
```

Proviso: notequal$(D_2,0) \wedge$ notequal$(D_2,0)$

(%o11) $\operatorname{span}\left(\begin{pmatrix} D_2 \\ -D_2-D_1 \\ 0 \end{pmatrix}\right)$

Se $q_1 = 0$ le singolarità di velocita si hanno per $v \in \operatorname{Im}\left\{\begin{pmatrix} -D_2 \\ D_2-D_1 \\ 0 \end{pmatrix}\right\}$.

Se q_2=π:

```
(%i12) Jq2:subst(q[2]=%pi,Jq);
```

(%o12) $\begin{pmatrix} -D_2\sin(q_1)-D_1\sin(q_1) & -D_2\sin(q_1) & 0 \\ D_2\cos(q_1)+D_1\cos(q_1) & D_2\cos(q_1) & 0 \\ 0 & 0 & 1 \end{pmatrix}$

```
(%i13) nullspace(Jq2);
```

Proviso: notequal$(-D_2\sin(q_1),0) \wedge$ notequal$(-D_2\sin(q_1),0)$

(%o13) $\operatorname{span}\left(\begin{pmatrix} -D_2\sin(q_1) \\ (D_2+D_1)\sin(q_1) \\ 0 \end{pmatrix}\right)$

```
(%i14)
```

Singolarità di Forza

```
(%i12) J:-transpose(J);
```

(%o12) $\begin{pmatrix} D_2\sin(q_2+q_1)+D_1\sin(q_1) & -D_2\cos(q_2+q_1)-D_1\cos(q_1) & 0 \\ D_2\sin(q_2+q_1) & -D_2\cos(q_2+q_1) & 0 \\ 0 & 0 & -1 \end{pmatrix}$

```
(%i13) dJ:trigreduce(expand(determinant(J)));
```

(%o13) $-D_1\,D_2\sin(q_2)$

```
(%i14) Jq:subst(q[2]=0,J);
```

(%o14) $$\begin{pmatrix} D_2 \sin(q_1) + D_1 \sin(q_1) & -D_2 \cos(q_1) - D_1 \cos(q_1) & 0 \\ D_2 \sin(q_1) & -D_2 \cos(q_1) & 0 \\ 0 & 0 & -1 \end{pmatrix}$$

```
(%i15) nullspace(Jq);
```

Proviso: notequal$((D_2 + D_1)\sin(q_1), 0) \wedge$ notequal$((-D_2 - D_1)\sin(q_1), 0)$

(%o15) $$\mathrm{span}\left(\begin{pmatrix} (-D_2 - D_1)\cos(q_1) \\ (-D_2 - D_1)\sin(q_1) \\ 0 \end{pmatrix}\right)$$

Se $q_1 \neq 0$ le singolarità di forza per $\tau \in \Im m\left\{\begin{pmatrix} (-D_2 - D_1)\cos(q_1) \\ (-D_2 - D_1)\sin(q_1) \\ 0 \end{pmatrix}\right\}$

Se $q_1 = 0$:

```
(%i16) Jq1:subst(q[1]=0,Jq);
```

(%o16) $$\begin{pmatrix} 0 & -D_2 - D_1 & 0 \\ 0 & -D_2 & 0 \\ 0 & 0 & -1 \end{pmatrix}$$

```
(%i17) nullspace(Jq1);
```

Proviso: notequal$(-D_2 - D_1, 0) \wedge$ notequal$(D_2 + D_1, 0)$

(%o17) $$\mathrm{span}\left(\begin{pmatrix} D_2 + D_1 \\ 0 \\ 0 \end{pmatrix}\right)$$

```
(%i18)
```

le singolarità di forza per $\tau \in \Im m\left\{\begin{pmatrix} D_2 + D_1 \\ 0 \\ 0 \end{pmatrix}\right\}$