Parametrizzazione di Cayley

La parametrizzazione di Cayley permette di esprimere, attraverso la matrice S antisimmetrica, la corrispettiva matrice di rotazione R e viceversa. La parametrizzazione di Cayley avviene tramite le seguenti relazioni:

$$(1) S \longrightarrow R = (I+S)(I-S)^{-1} \Longrightarrow R \quad \text{è di rotazione}$$

$$(2) R \longrightarrow S = (R+I)^{-1}(R-I) \Longrightarrow S \quad \text{è una matrice antisimmetrica}$$

In particolare, dato un asse di rotazione v occorre prima calcolare la matrice antisimmetrica S ed in seguito applicare la (1) per ottenere la corrispondente matrice di rotazione.

Alternativamente, data una matrice di rotazione R occorre applicare (2) per ottenere la matrice antisimmetrica corrispondente e, in seguito, selezionare all'interno della matrice S gli elementi dell'asse di rotazione v. Quest'ultimi corrispondono a:

$$S = \begin{pmatrix} 0 & -c & b \\ c & 0 & -a \\ -b & a & 0 \end{pmatrix} \longrightarrow v = \begin{pmatrix} a \\ b \\ c \end{pmatrix}$$

Ottenuto il vettore v è possibile calcolare il versore e l'angolo di rotazione. Infatti:

$$v = \frac{v}{\|v\|}\|v\| \quad \text{in cui} \quad \frac{v}{\|v\|} := \text{versore di rotazione} \quad \|v\| := \text{angolo di rotazione}$$

N.B.: $\|v\|$ è la norma-2 del vettore v

Per calcolare correttamente l'angolo di rotazione data la norma del vettore v, poniamo v=α. Quindi, sussistono le seguenti relazioni:

$$\alpha = \frac{\sin(\theta)}{1+\cos(\theta)} \longrightarrow \alpha^2 = \frac{1-\cos(\theta)}{1+\cos(\theta)}$$

$$(i) \qquad \cos(\theta) = \frac{1-\alpha^2}{1+\alpha^2} \equiv \frac{1-\|v\|^2}{1+\|v\|^2}$$

Data la pluralità delle soluzioni occorre calcolare anche il $\sin(\theta)$:

$$\sin(\theta)^2 = 1 - \cos(\theta)^2$$

Sostituendo (i):

$$\sin(\theta) = \pm\frac{2\,\alpha}{1+\alpha^2}$$

In cui il segno di sin(ϑ) deve soddisfare la condizione in cui $\|v\|$>0 cioé: $\frac{\sin(\theta)}{1+\cos(\theta)} > 0$.

Dopodiché l'angolo è univocamente identificato tramite la funzione atan2:

$$(\text{ii})\theta = \text{atan2}\left(\pm\frac{2\,\alpha}{1+\alpha^2}, \frac{1-\alpha^2}{1+\alpha^2}\right) = \text{atan2}(\sin(\theta), \cos(\theta))$$

```
Maxima 5.44.0 http://maxima.sourceforge.net
using Lisp SBCL 2.0.0
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
```

```
The function bug_report() provides bug reporting information.
```

La funzione isRotation verifica se la matrice in input M è di rotazione o meno. Restituisce 1 in caso positvo, altrimenti errore.

```
(%i1) isRotation(M):=block([res],
                           I:ident(3),
                           MMT:trigsimp(expand(M.transpose(M))),
                           detM:trigsimp(expand(determinant(M))),

                           if MMT=I  and detM=1
                              then(

                                   return(res:1)
                                   )

                           else(

                                 res: "R is not rotation matrix"
                                 )
                           )
```

(%o1) isRotation(M) := **block** ([res], I: ident(3), MMT: trigsimp(expand($M \cdot$ transpose(M))), detM: trigsimp(expand(determinant(M))), **if** MMT $= I \wedge$ detM $= 1$ **then** return(res: 1) **else** res: R is not rotation matrix)

La funzione skewMatric(x) prende in input una vettore x e ne costruisce l'antisimetrica

```
(%i2) skewMatrix(x):=block([res],
                           M:ident(3),
                           for i:1 thru 3 do
                           (
                           for j:1 thru 3 do
                              (
                                if i=j
                                   then M[i][j]:0
                                elseif j>i
                                   then (
                                  temp:(-1)^(j-i)*x[3-remainder(i+j,3)],
                                         M[i][j]:temp,
                                         M[j][i]:-temp

                                          )
                                )
                            ),
                            res:M
                            )
```

(%o2) skewMatrix(x) := **block** ([res], M: ident(3), **for** i **thru** 3 **do for** j **thru** 3 **do if** $i = j$ **then** $(M_i)_j$: 0 **elseif** $j > i$ **then** (temp: $(-1)^{j-i}\, x_{3-\mathrm{remainder}(i+j,3)}$, $(M_i)_j$: temp, $(M_j)_i$: $-$temp), res: M)

La funzione degreeVector(x) prende in input un vettore x e ne calcola l'angolo come descritto nella procedura (ii) scegliendo opportunamente il segno del $\sin(\theta)$

```
(%i3) degreeVector(x):=block([res],
                              vNorm:x.x,
                                cosTheta:(1-vNorm)/(1+vNorm),
                                sinTheta:(2*sqrt(vNorm))/(1+vNorm),

                                if sinTheta/(1+cosTheta)>0
                                   then (print(atan2(sinTheta,cosTheta)),
      degree:atan2(sinTheta,cosTheta))
                                else (degree:atan2(-sinTheta,cosTheta)),
                                res:degree
                                )
```

(%o3) $\text{degreeVector}(x) := \textbf{block}\left([\text{res}], \text{vNorm}\colon x\cdot x, \text{cosTheta}\colon \frac{1-\text{vNorm}}{1+\text{vNorm}}, \text{sinTheta}\colon \frac{2\sqrt{\text{vNorm}}}{1+\text{vNorm}}, \textbf{if } \frac{\text{sinTheta}}{1+\text{cosTheta}} > 0 \textbf{ then } (\text{print}(\text{atan2}(\text{sinTheta}, \text{cosTheta})), \text{degree}\colon \text{atan2}(\text{sinTheta}, \text{cosTheta})) \textbf{ else } \text{degree}\colon \text{atan2}(-\text{sinTheta}, \text{cosTheta}), \text{res}\colon \text{degree}\right)$

La funzione cayleyRotation(a) prende in input un vettore a, ne calcola l'antisimmetrica e resistuisce in output la corrispondente matrice di rotazione R attraverso la parametrizzazione di Cayley(1). In output vi sono per completezza anche la corrispettiva matrice antisimmetrica, asse di rotazione e angolo di rotazione.

```
(%i4) cayleyRotation(a):=block([res],
                                S:skewMatrix(a),
                                I:ident(3),
                                R:(I+S).invert(I-S),
                                degree:degreeVector(a),
                                print("Matrice di rotazione, Matrice
      antisimmetrica, asse, angolo"),
                                res:[facsum(expand(R)),S,a,degree])
```

(%o4) $\text{cayleyRotation}(a) := \textbf{block}\ ([\text{res}], S\colon \text{skewMatrix}(a), I\colon \text{ident}(3), R\colon (I+S)\cdot \text{invert}(I-S), \text{degree}\colon \text{degreeVector}(a), \text{print}(\text{Matrice di rotazione, Matrice antisimmetrica, asse, angolo }), \text{res}\colon [\text{facsum}(\text{expand}(R)), S, a, \text{degree}])$

La funzione cayleySkewMatrix(R) prende in input una matrice R e, tramite la parametrizzazione di Cayley, restituisce in output la corrispondente matrice antisimmetrica S.
In particolare, per completezza viene restituito un array contenente la matrice di rotazione, la matrice antisimmetrica, asse e angolo di rotazione.

```
(%i5) cayleySkewMatrix(R):=block([res],
                                  isRot:isRotation(R),
                                  if isRot=1 then(
                                     I:ident(3),
                                     S:invert(R+I).(R-I),
                                     v:[S[3][2],S[1][3],S[2][1]],
                                     degree:degreeVector(v),
                                     print("Matrice di rotazione, Matrice
      antisimmetrica, asse, angolo"),
                                     res:[R,S,v,degree]

                                  )
                                  else res:"Matrix is not rotation"
                                  )
```

(%o5) cayleySkewMatrix(R) := **block** ([res], isRot: isRotation(R), **if** isRot = 1 **then** (I: ident(3), S: invert$(R+I)\cdot(R-I)$, v: $[(S_3)_2, (S_1)_3, (S_2)_1]$, degree: degreeVector$(v)$, print(Matrice di rotazione, Matrice antisimmetrica, asse, angolo), res: $[R, S, v, \text{degree}]$) **else** res: Matrix is not rotation)

Asse di rotazione versore lungo l'asse x

(%i6) `v:[1,0,0]`

(%o6) $[1,0,0]$

(%i7) `a:cayleyRotation(v)`

$\frac{\pi}{2}$

Matrice di rotazione, Matrice antisimmetrica, asse, angolo

(%o7) $$\left[\begin{pmatrix}1 & 0 & 0\\ 0 & 0 & -1\\ 0 & 1 & 0\end{pmatrix}, \begin{pmatrix}0 & 0 & 0\\ 0 & 0 & -1\\ 0 & 1 & 0\end{pmatrix}, [1,0,0], \frac{\pi}{2}\right]$$

La prima e la seconda parametrizzazione di Cayley restituiscono gli stessi risultati.

(%i8) `RR:a[1]`

(%o8) $$\begin{pmatrix}1 & 0 & 0\\ 0 & 0 & -1\\ 0 & 1 & 0\end{pmatrix}$$

(%i9) `cayleySkewMatrix(a[1])`

$\frac{\pi}{2}$

Matrice di rotazione, Matrice antisimmetrica, asse, angolo

(%o9) $$\left[\begin{pmatrix}1 & 0 & 0\\ 0 & 0 & -1\\ 0 & 1 & 0\end{pmatrix}, \begin{pmatrix}0 & 0 & 0\\ 0 & 0 & -1\\ 0 & 1 & 0\end{pmatrix}, [1,0,0], \frac{\pi}{2}\right]$$

Parametrizzazione di Cayley simbolica:

(%i10) `v:[alpha,beta,%gamma];`

(%o10) $[\alpha, \beta, \gamma]$

(%i11) `a:cayleyRotation(v);`

$$\operatorname{atan2}\left(\frac{2\sqrt{\beta^2+\alpha^2+\gamma^2}}{\beta^2+\alpha^2+\gamma^2+1}, \frac{-\beta^2-\alpha^2-\gamma^2+1}{\beta^2+\alpha^2+\gamma^2+1}\right)$$

Matrice di rotazione, Matrice antisimmetrica, asse, angolo

(%o11) $$\left[\begin{pmatrix}-\frac{\beta^2-\alpha^2+\gamma^2-1}{\beta^2+\alpha^2+\gamma^2+1} & \frac{2(\alpha\beta-\gamma)}{\beta^2+\alpha^2+\gamma^2+1} & \frac{2(\beta+\gamma\alpha)}{\beta^2+\alpha^2+\gamma^2+1}\\ \frac{2(\alpha\beta+\gamma)}{\beta^2+\alpha^2+\gamma^2+1} & \frac{\beta^2-\alpha^2-\gamma^2+1}{\beta^2+\alpha^2+\gamma^2+1} & \frac{2(\gamma\beta-\alpha)}{\beta^2+\alpha^2+\gamma^2+1}\\ -\frac{2(\beta-\gamma\alpha)}{\beta^2+\alpha^2+\gamma^2+1} & \frac{2(\gamma\beta+\alpha)}{\beta^2+\alpha^2+\gamma^2+1} & -\frac{\beta^2+\alpha^2-\gamma^2-1}{\beta^2+\alpha^2+\gamma^2+1}\end{pmatrix}, \begin{pmatrix}0 & -\gamma & \beta\\ \gamma & 0 & -\alpha\\ -\beta & \alpha & 0\end{pmatrix}, [\alpha,\beta,\gamma], \operatorname{atan2}\left(\frac{2\sqrt{\beta^2+\alpha^2+\gamma^2}}{\beta^2+\alpha^2+\gamma^2+1}, \frac{-\beta^2-\alpha^2-\gamma^2+1}{\beta^2+\alpha^2+\gamma^2+1}\right)\right]$$

(%i12)