Cinematica diretta Robot Antropomorfo

N.B.: **le grandezze diverse da quelle di giunto q_i sono L_i, D_i. Esse sono rispettivamente la distanza tra i sistemi di riferimento R_i e R_{i+1} nelle operazioni della matrice avvitamento $A_z(\theta, d)$ e $A_x(\alpha, a)$.**



	ϑ	d	α	a
1	q_1	L_1	$\frac{\pi}{2}$	0
2	q_2	0	0	L_2
3	q_3	0	0	L_3

Tabella 1.

Funzioni ausiliarie:

```
(%i1) inverseLaplace(SI,theta):=block([res],
                              M:SI,
                              MC:SI,
                              for i:1 thru 3 do(
                                for j:1 thru 3 do
                                   (
                                     aC:M[i,j],
                                     b:ilt(aC,s,theta),
                                     MC[i,j]:b
                                    )
                                 ),
                              res:MC

                           )
```

(%o1) inverseLaplace(SI, ϑ) := **block** ([res], M: SI, MC: SI, **for** i **thru** 3 **do for** j **thru** 3 **do** (aC: $M_{i,j}$, b: ilt(aC, s, ϑ), MC$_{i,j}$: b), res: MC)

(%i2)
```
rotLaplace(k,theta):=block([res],
                    S:ident(3),
                    I:ident(3),
                  for i:1 thru 3 do
                    (
                    for j:1 thru 3 do
                       (
                         if i=j
                            then S[i][j]:0
                         elseif j>i
                            then (
                           temp:(-1)^(j-i)*k[3-remainder(i+j,3)],
                                  S[i][j]:temp,
                                  S[j][i]:-temp

                                   )
                         )
                     ),
                    res:inverseLaplace(invert(s*I-S),theta)


                   )
```

(%o2) $\mathrm{rotLaplace}(k,\vartheta):=\mathbf{block}\ ([\mathrm{res}], S\colon \mathrm{ident}(3), I\colon \mathrm{ident}(3),$ $\mathbf{for}\ i\ \mathbf{thru}\ 3\ \mathbf{do}\ \mathbf{for}\ j\ \mathbf{thru}\ 3\ \mathbf{do}\ \mathbf{if}\ i=j\ \mathbf{then}\ (S_i)_j\colon 0\ \mathbf{elseif}\ j>i\ \mathbf{then}\ (\mathrm{temp}\colon$ $(-1)^{j-i}\,k_{3-\mathrm{remainder}(i+j,3)}, (S_i)_j\colon \mathrm{temp}, (S_j)_i\colon -\mathrm{temp}), \mathrm{res}\colon \mathrm{inverseLaplace}(\mathrm{invert}(s\,I-S),\vartheta))$

(%i3)
```
Av(v,theta,d):=block([res],
                    Trot:rotLaplace(v,theta),
                    row:matrix([0,0,0,1]),
                    Atemp:addcol(Trot,d*transpose(v)),
                    A:addrow(Atemp,row),
                    res:A
                    )
```

(%o3) $\mathrm{Av}(v,\vartheta,d):=\mathbf{block}\ ([\mathrm{res}], \mathrm{Trot}\colon \mathrm{rotLaplace}(v,\vartheta), \mathrm{row}\colon \begin{pmatrix} 0 & 0 & 0 & 1 \end{pmatrix}, \mathrm{Atemp}\colon \mathrm{addcol}(\mathrm{Trot},$ $d\,\mathrm{transpose}(v)), A\colon \mathrm{addrow}(\mathrm{Atemp},\mathrm{row}), \mathrm{res}\colon A)$

(%i4)
```
Q(theta,d,alpha,a):=block([res],
                         tempMat:Av([0,0,1],theta,d).Av([1,0,0],alpha,a),
                         Qtrasf:zeromatrix(4,4),
                         for i:1 thru 4 do
                    (
                    for j:1 thru 4 do
                       (
                         Qtrasf[i][j]:trigreduce(tempMat[i][j])
                                   )
                       ),

                         res:Qtrasf
                         )
```

(%o4) $Q(\vartheta,d,\alpha,a):=\mathbf{block}\ ([\mathrm{res}], \mathrm{tempMat}\colon \mathrm{Av}([0,0,1],\vartheta,d)\cdot\mathrm{Av}([1,0,0],\alpha,a), \mathrm{Qtrasf}\colon$ $\mathrm{zeromatrix}(4,4), \mathbf{for}\ i\ \mathbf{thru}\ 4\ \mathbf{do}\ \mathbf{for}\ j\ \mathbf{thru}\ 4\ \mathbf{do}\ (\mathrm{Qtrasf}_i)_j\colon \mathrm{trigreduce}((\mathrm{tempMat}_i)_j), \mathrm{res}\colon \mathrm{Qtrasf})$

(%i5) `let(sin(q[1]), s[1]);`

(%o5) $\sin(q_1) \longrightarrow s_1$

(%i6) `let(sin(q[2]), s[2]);`

(%o6) $\sin(q_2) \longrightarrow s_2$

(%i7) `let(cos(q[1]),c[1]);`

(%o7) $\cos(q_1) \longrightarrow c_1$

(%i8) `let(cos(q[2]),c[2]);`

(%o8) $\cos(q_2) \longrightarrow c_2$

(%i9) `let(sin(q[1]+q[2]), s[12]);`

(%o9) $\sin(q_2 + q_1) \longrightarrow s_{12}$

(%i10) `let(cos(q[1]+q[2]),c[12]);`

(%o10) $\cos(q_2 + q_1) \longrightarrow c_{12}$

(%i11) `let(sin(q[2]+q[3]),s[23]);`

(%o11) $\sin(q_3 + q_2) \longrightarrow s_{23}$

(%i12) `let(cos(q[2]+q[3]),c[23]);`

(%o12) $\cos(q_3 + q_2) \longrightarrow c_{23}$

(%i13) `let(sin(q[1]+q[3]),s[23]);`

(%o13) $\sin(q_3 + q_1) \longrightarrow s_{23}$

(%i14) `let(cos(q[1]+q[3]),c[13]);`

(%o14) $\cos(q_3 + q_1) \longrightarrow c_{13}$

(%i15) `let(sin(q[3]),s[3]);`

(%o15) $\sin(q_3) \longrightarrow s_3$

(%i16) `let(cos(q[3]),q[3]);`

(%o16) $\cos(q_3) \longrightarrow q_3$

(%i17)

Cinematica diretta:

(%i17)
```
Q[antropomorfo](q1,q2,q3,L1,L2,L3):=Q(q1,L1,%pi/2,0).
                        trigreduce(trigexpand(Q(q2,0,0,L2).
                        Q(q3,0,0,L3)));
```

(%o17) $Q_{\text{antropomorfo}}(q1, q2, q3, L1, L2, L3) := Q\left(q1, L1, \frac{\pi}{2}, 0\right) \cdot \text{trigreduce}(\text{trigexpand}(Q(q2, 0, 0, L2) \cdot Q(q3, 0, 0, L3)))$

(%i18) `Qantropomorfo:Q[antropomorfo](q[1],q[2],q[3],L[1],L[2],L[3]);`

(%o18) $(\cos(q_1)\cos(q_3+q_2), -\cos(q_1)\sin(q_3+q_2), \sin(q_1), \cos(q_1)(L_3\cos(q_3+q_2) + L_2\cos(q_2)); \sin(q_1)\cos(q_3+q_2), -\sin(q_1)\sin(q_3+q_2), -\cos(q_1), \sin(q_1)(L_3\cos(q_3+q_2) + L_2\cos(q_2)); \sin(q_3+q_2), \cos(q_3+q_2), 0, L_3\sin(q_3+q_2) + L_2\sin(q_2) + L_1; 0, 0, 0, 1)$

(%i19) `letsimp(Qantropomorfo);`

(%o19)
$$\begin{pmatrix} c_1\,c_{23} & -c_1\,s_{23} & s_1 & c_1\,L_3\,c_{23} + c_1\,L_2\,c_2 \\ s_1\,c_{23} & -s_1\,s_{23} & -c_1 & s_1\,L_3\,c_{23} + s_1\,L_2\,c_2 \\ s_{23} & c_{23} & 0 & L_3\,s_{23} + L_2\,s_2 + L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

(%i20)

Cinematica inversa robot antropomorfo

Al fine di risolvere il problema di cinematica inversa del robot antropomorfo occorre risolevere il problema di posizione ed orientamente inverso.Inizialmente individuare lo spazio di lavoro, le soluzioni generiche, singolari ed infine le variabili di giunto q_i ed in seguito determinare l'orientamento del robot. Dalla cinematica diretta sappiamo che:

$$Q_{\text{antropomorfo}} = \begin{pmatrix} c_1 c_{23} & -c_1 s_{23} & s_1 & c_1 L_3 c_{23} + c_1 L_2 c_2 \\ s_1 c_{23} & -s_1 s_{23} & -c_1 & s_1 L_3 c_{23} + s_1 L_2 c_2 \\ s_{23} & c_{23} & 0 & L_3 s_{23} + L_2 s_2 + L_1 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Cinematica inversa di posizione

$$\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} c_1 L_3 c_{23} + c_1 L_2 c_2 \\ s_1 L_3 c_{23} + s_1 L_2 c_2 \\ L_3 s_{23} + L_2 s_2 + L_1 \end{pmatrix}$$

$$\begin{cases} x = c_1 L_3 c_{23} + c_1 L_2 c_2 \\ y = s_1 L_3 c_{23} + s_1 L_2 c_2 \end{cases} \rightarrow \begin{cases} x = c_1 (L_3 c_{23} + L_2 c_2) \\ y = s_1 (L_3 c_{23} + L_2 c_2) \end{cases} \rightarrow \begin{cases} x^2 = c_1^2 (L_3 c_{23} + L_2 c_2)^2 \\ y^2 = s_1^2 (L_3 c_{23} + L_2 c_2)^2 \end{cases}$$

$$x^2 + y^2 = (L_3 c_{23} + L_2 c_2)^2 \rightarrow L_3 c_{23} + L_2 c_2 = \pm\sqrt{x^2 + y^2}$$

$$\begin{cases} L_3 c_{23} + L_2 c_2 = \pm\sqrt{x^2 + y^2} \\ L_3 s_{23} + L_2 s_2 = z - L_1 \end{cases}$$

è possibile riscrivere le ultime due equazioni nel seguente modo:

$$\begin{pmatrix} c_{23} & -s_{23} \\ s_{23} & c_{23} \end{pmatrix} \begin{pmatrix} L_3 \\ 0 \end{pmatrix} + \begin{pmatrix} c_2 & -s_2 \\ s_2 & c_2 \end{pmatrix} \begin{pmatrix} L_2 \\ 0 \end{pmatrix} = \begin{pmatrix} \pm\sqrt{x^2 + y^2} \\ z - L_1 \end{pmatrix}$$

In particolare $R_{23} = \begin{pmatrix} c_{23} & -s_{23} \\ s_{23} & c_{23} \end{pmatrix}$ è una matrice di rotazione nel piano quindi equivale a $R_2 R_3$:

$$R_2 \left(R_3 \begin{pmatrix} L_3 \\ 0 \end{pmatrix} + \begin{pmatrix} L_2 \\ 0 \end{pmatrix} \right) = \begin{pmatrix} \pm\sqrt{x^2 + y^2} \\ z - L_1 \end{pmatrix}$$

Poiché R_2 è una matrice di rotazione i termini $R_3 \begin{pmatrix} L_3 \\ 0 \end{pmatrix} + \begin{pmatrix} L_2 \\ 0 \end{pmatrix}, \begin{pmatrix} \pm\sqrt{x^2 + y^2} \\ z - L_1 \end{pmatrix}$ devono avere la stessa norma:

$$\left(\begin{pmatrix} L_3 & 0 \end{pmatrix} R_3^T + \begin{pmatrix} L_2 & 0 \end{pmatrix} \right) \left(R_3 \begin{pmatrix} L_3 \\ 0 \end{pmatrix} + \begin{pmatrix} L_2 \\ 0 \end{pmatrix} \right) =$$

$$= \begin{pmatrix} L_3 & 0 \end{pmatrix} R_3^T R_3 \begin{pmatrix} L_3 \\ 0 \end{pmatrix} + \begin{pmatrix} L_2 & 0 \end{pmatrix} \begin{pmatrix} L_2 \\ 0 \end{pmatrix} + 2 \begin{pmatrix} L_2 & 0 \end{pmatrix} R_3 \begin{pmatrix} L_3 \\ 0 \end{pmatrix} =$$

$$= L_3^2 + L_2^2 + 2\, L_2 L_3 c_3$$

Quindi:

$$x^2 + y^2 + (z - L_1)^2 = L_3^2 + L_2^2 + 2\, L_2 L_3 c_3$$

$$c_3 = \frac{x^2 + y^2 + (z - L_1)^2 - L_3^2 - L_2^2}{2\, L_2 L_3}$$

Imponendo la condizione che $-1 \leqslant c_3 \leqslant 1$:

$$-1 \leqslant \frac{x^2 + y^2 + (z - L_1)^2 - L_3^2 - L_2^2}{2\, L_2 L_3} \leqslant 1$$

Otteniamo l'espressione dello spazio operativo:

$$(L_3 - L_2)^2 \leqslant x^2 + y^2 + (z - L_1)^2 \leqslant (L_3 + L_2)^2$$

che rappresenta una sfera cava di centro $\begin{pmatrix} 0 \\ 0 \\ L_1 \end{pmatrix}$ e raggio $|L_2 - L_3| \leqslant r \leqslant L_2 + L_3$.

Per determinare la variabile di giunto q_3:

$$c_3 = \frac{x^2 + y^2 + (z - L_1)^2 - L_3^2 - L_2^2}{2\, L_2 L_3}$$

$$s_3 = \pm\sqrt{1 - c_3}$$

$$q_3 = \text{atan2}(\pm s_3, c_3)$$

A questo punto il termine $R_3\begin{pmatrix} L_3 \\ 0 \end{pmatrix} + \begin{pmatrix} L_2 \\ 0 \end{pmatrix}$ è una quantità nota detta A_1, A_2. In aggiunta, i termni $\begin{pmatrix} \pm\sqrt{x^2 + y^2} \\ z - L_1 \end{pmatrix}$ li definiamo come B_1, B_2 ottendendo:

$$\begin{pmatrix} c_2 & -s_2 \\ s_2 & c_2 \end{pmatrix} \begin{pmatrix} A_1 \\ A_2 \end{pmatrix} = \begin{pmatrix} \pm B_1 \\ B_2 \end{pmatrix}$$

$$\begin{pmatrix} A_1 & -A_2 \\ A_2 & A_1 \end{pmatrix} \begin{pmatrix} c_2 \\ s_2 \end{pmatrix} = \begin{pmatrix} \pm B_1 \\ B_2 \end{pmatrix}$$

$$\det\left(\begin{pmatrix} A_1 & -A_2 \\ A_2 & A_1 \end{pmatrix}\right) = A_1^2 + A_2^2 \neq 0 \rightarrow \text{è possibile effettuare} \quad l'\text{inversa}$$

$$\begin{pmatrix} c_2 \\ s_2 \end{pmatrix} = \frac{1}{A_1^2 + A_2^2} \begin{pmatrix} A_1 & A_2 \\ -A_2 & A_1 \end{pmatrix} \begin{pmatrix} \pm B_1 \\ B_2 \end{pmatrix}$$

$$\begin{pmatrix} c_2 \\ s_2 \end{pmatrix} = \begin{pmatrix} \frac{\pm A_1 B_1 + A_2 B_2}{A_1^2 + A_2^2} \\ \frac{A_1 B_2 \mp A_2 B_1}{A_1^2 + A_2^2} \end{pmatrix}$$

$$q_2 = \text{atan2}(s_2, c_2) = \text{atan2}\left(\frac{A_1 B_2 \mp A_2 B_1}{A_1^2 + A_2^2}, \frac{\pm A_1 B_1 + A_2 B_2}{A_1^2 + A_2^2}\right) =$$

$$= \text{atan2}(A_1 B_2 \mp A_2 B_1, \pm A_1 B_1 + A_2 B_2)$$

Per la variabile di giunto q_1, si ricorda che:

$$\begin{cases} x = c_1\,(L_3\, c_{23} + L_2\, c_2) \\ y = s_1\,(L_3\, c_{23} + L_2\, c_2) \end{cases}$$

Poiché $(L_3\, c_{23} + L_2\, c_2)$ è una quantità nota:

$$\begin{cases} c_1 = \frac{x}{(L_3\, c_{23} + L_2\, c_2)} \\ s_1 = \frac{y}{(L_3\, c_{23} + L_2\, c_2)} \end{cases}$$

In conclusione:

$$q_1 = \text{atan2}(s_1, c_1) = \text{atan2}\left(\frac{y}{(L_3\, c_{23} + L_2\, c_2)}, \frac{x}{(L_3\, c_{23} + L_2\, c_2)}\right)$$

Orientamento inverso

La risoluzione del problema di orientamento inverso si basa sulla scelta di una terna di Eulero o di una terna nautica in condizioni non singolari se possibile.

$$R_{\text{antorpomorfo}}=\begin{pmatrix} c_1\,c_{23} & -c_1\,s_{23} & s_1\\ s_1\,c_{23} & -s_1\,s_{23} & -c_1\\ s_{23} & c_{23} & 0 \end{pmatrix}$$

$$R_{\text{zyx}}=\begin{pmatrix} c_y c_z & \cdots\cdots\cdots & \\ c_y s_z & \cdots\cdots\cdots & \\ -s_y & s_x c_y & c_x c_y \end{pmatrix}$$

$$-s_y=s_{23}\neq\pm1\rightarrow q_2+q_3=\begin{cases} \pm\frac{\pi}{2}\rightarrow \text{soluzione singolare}\\ \text{altrimenti}\rightarrow \text{soluzine regolare}\end{cases}$$

$$\begin{cases} s_y=-s_{23}\\ c_y=\pm\sqrt{1-s_y}=\pm c_{23}\end{cases}\rightarrow\phi_y=\text{atan2}(-s_{23},c_{23})=\begin{cases} -(q_2+q_3)\\ \pi+(q_2+q_3)\end{cases}$$

$$\begin{cases} s_x c_y=c_{23}\\ c_x c_y=0\end{cases}\rightarrow\begin{cases} \pm s_x c_{23}=c_{23}\\ \pm c_x c_{23}=0\end{cases}\rightarrow\begin{cases} s_x=\pm1\\ c_x=0\end{cases}\rightarrow\phi_x=\text{atan2}(\pm1,0)=\begin{cases} \frac{\pi}{2}\\ -\frac{\pi}{2}\end{cases}$$

$$\begin{cases} c_y c_z=c_1 c_{23}\\ c_y s_z=s_1 c_{23}\end{cases}\rightarrow\begin{cases} c_z=\pm c_1\\ s_z=\pm s_1\end{cases}\rightarrow\phi_z=\text{atan2}(\pm s_1,\pm c_1)=\begin{cases} q_1\\ q_1+\pi\end{cases}$$

Riassumendo:

$$\begin{pmatrix} \frac{\pi}{2}\\ -(q_2+q_3)\\ q_1\end{pmatrix},\begin{pmatrix} -\frac{\pi}{2}\\ \pi+q_2+q_3\\ q_1+\pi\end{pmatrix}$$

In alternativa, tramite la scelta di una terna di Eulero:

$$R_{\text{zyz}}=\begin{pmatrix} \cdots & \cdots & \cos(\alpha)\sin(\beta)\\ \cdots & \cdots & \sin(\alpha)\sin(\beta)\\ -\sin(\beta)\cos(\gamma) & \sin(\beta)\sin(\gamma) & \cos(\beta)\end{pmatrix}$$

$$\cos(\beta)=0\rightarrow\sin(\beta)=\pm1\rightarrow\beta=\text{atan2}(\pm1,0)=\begin{cases} \frac{\pi}{2}\\ -\frac{\pi}{2}\end{cases}$$

$$\begin{cases} \sin(\beta)\sin(\gamma)=c_{23}\\ -\sin(\beta)\cos(\gamma)=s_{23}\end{cases}\rightarrow\begin{cases} \sin(\gamma)=\pm c_{23}\\ \cos(\gamma)=\mp s_{23}\end{cases}\rightarrow\gamma=\text{atan2}(\pm c_{23},\mp s_{23})$$

$$\gamma=\text{atan2}(\pm c_{23},\mp s_{23})=\begin{cases} q_2+q_3+\frac{\pi}{2}\\ q_2+q_3-\frac{\pi}{2}\end{cases}$$

$$\begin{cases} \cos(\alpha)\sin(\beta)=s_1\\ \sin(\alpha)\sin(\beta)=-c_1\end{cases}\rightarrow\begin{cases} \cos(\alpha)=\pm s_1\\ \sin(\alpha)\mp c_1\end{cases}\rightarrow\alpha=\text{atan2}(\mp c_1,\pm s_1)=\begin{cases} q_1-\frac{\pi}{2}\\ q_1+\frac{\pi}{2}\end{cases}$$

Riassumendo:

$$\begin{pmatrix} q_1-\frac{\pi}{2}\\ \frac{\pi}{2}\\ q_2+q_3+\frac{\pi}{2}\end{pmatrix},\begin{pmatrix} q_1+\frac{\pi}{2}\\ -\frac{\pi}{2}\\ q_2+q_3-\frac{\pi}{2}\end{pmatrix}$$

(%i17)
```
isRotation(M):=block([MC,res],
                        I:ident(3),
                        MC:ident(3),
                        for i:1 thru 3 do
                        (
                        for j:1 thru 3 do
                           (
                              MC[i][j]:M[i][j]
                              )
                           ),
                        MMT:trigsimp(expand(MC.transpose(MC))),
                        detM:trigsimp(expand(determinant(MC))),

                        if MMT=I  and detM=1
                           then(

                                return(res:1)
                                )

                        else(

                              res: "R is not rotation matrix"
                              )
                        )
```

(%o17) isRotation(M) := **block** ([MC, res], I: ident(3), MC: ident(3), **for** i **thru** 3 **do for** j **thru** 3 **do** $(\mathrm{MC}_i)_j$: $(M_i)_j$, MMT: trigsimp(expand(MC $\cdot$ transpose(MC))), detM: trigsimp(expand(determinant(MC))), **if** MMT $= I \wedge$ detM $= 1$ **then** return(res: 1) **else** res: R is not rotation matrix)

(%i18)
```
skewMatrix(x):=block([res],
                        S:ident(3),
                        for i:1 thru 3 do
                        (
                        for j:1 thru 3 do
                           (
                             if i=j
                                then S[i][j]:0
                             elseif j>i
                                then (
                               temp:(-1)^(j-i)*x[3-remainder(i+j,3)],
                                      S[i][j]:temp,
                                      S[j][i]:-temp

                                       )
                             )
                         ),
                         res:S
                         )
```

(%o18) skewMatrix(x) := **block** ([res], S: ident(3), **for** i **thru** 3 **do for** j **thru** 3 **do if** $i = j$ **then** $(S_i)_j$: 0 **elseif** $j > i$ **then** (temp: $(-1)^{j-i}\, x_{3-\mathrm{remainder}(i+j,3)}$, $(S_i)_j$: temp, $(S_j)_i$: $-$temp), res: S)

(%i19)
```
rodriguez(y,arg):=block([res],
                        I:ident(3),
                        S:skewMatrix(y),
                        res:I+S.S*(1-cos(arg))+S*sin(arg)
                        )
```

(%o19) $\mathrm{rodriguez}(y,\mathrm{arg}) := \textbf{block}\ ([\mathrm{res}], I\colon \mathrm{ident}(3), S\colon \mathrm{skewMatrix}(y), \mathrm{res}\colon I + S\cdot S\,(1-\cos(\mathrm{arg})) + S\sin(\mathrm{arg}))$

(%i20)
```
calculate(x,y,z,L1,L2,L3):=block(
[c1,s1,c2,s2,c3,s3,res,A,B,q1,q2,q3],
condition: x^(2)+y^(2)+(z-L1)^2,
if(condition>(L2+L3)^2 or condition<(L2-L3)^2 or (x=0 and y=0)) then
             (error("La soluzione è singolare")),
      c3:trigsimp(ratsimp((condition-L3^(2)-L2^(2))/(2*L2*L3))),
      s3:trigsimp(ratsimp(sqrt(1-c3^2))),
      q3:atan2(s3,c3),
      B:[ratsimp(sqrt(x^(2)+y^(2))),ratsimp(z-L1)],
      A:[trigsimp(ratsimp(cos(q3)*L3+L2)),
         abs(trigsimp(ratsimp(sin(q3)*L3)))],
      c2:[A[1]*B[1]+A[2]*B[2],
         -A[1]*B[1]+A[2]*B[2],
         A[1]*B[1]-A[2]*B[2],
         -A[1]*B[1]-A[2]*B[2]],
      s2:[-A[2]*B[1]+A[1]*B[2],
          A[2]*B[1]+A[1]*B[2],
          A[2]*B[1]+A[1]*B[2],
          -A[2]*B[1]+A[1]*B[2]],
      q2:[atan2(ratsimp(s2[1]),ratsimp(c2[1])),
          atan2(ratsimp(s2[2]),ratsimp(c2[2])),
          atan2(ratsimp(s2[3]),ratsimp(c2[3])),
          atan2(ratsimp(s2[4]),ratsimp(c2[4]))],
       den:[L3*cos(q3+q2[1])+L2*cos(q2[1]),
         L3*cos(q3+q2[2])+L2*cos(q2[2]),
         L3*cos(-q3+q2[3])+L2*cos(q2[3]),
         L3*cos(-q3+q2[4])+L2*cos(q2[4])],
       c1:[ratsimp(x/den[1]),
           ratsimp(x/den[2]),
           ratsimp(x/den[3]),
           ratsimp(x/den[4])],
       s1:[ratsimp(y/den[1]),
           ratsimp(y/den[2]),
           ratsimp(y/den[3]),
           ratsimp(y/den[4])],
       q1:[atan2(s1[1],c1[1]),
          atan2(s1[2],c1[2]),
          atan2(s1[3],c1[3]),
          atan2(s1[4],c1[4])],

       res:[[q1[1],q2[1],q3],
            [q1[2],q2[2],q3],
            [q1[3],q2[3],-q3],
            [q1[4],q2[4],-q3]]
)
```

(%o20) $\mathrm{calculate}(x,y,z,L1,L2,L3) := \textbf{block}\ \Big([c1, s1, c2, s2, c3, s3, \mathrm{res}, A, B, q1, q2, q3],$

condition: $x^2+y^2+(z-L1)^2$, **if** condition $>(L2+L3)^2 \vee$ condition $<(L2-L3)^2 \vee x=0 \wedge y=0$ **then** error(La soluzione è singolare), $c3$: trigsimp$\left(\text{ratsimp}\left(\frac{\text{condition}-L3^2-L2^2}{2\,L2\,L3}\right)\right)$, $s3$: trigsimp$\left(\text{ratsimp}\left(\sqrt{1-c3^2}\right)\right)$, $q3$: atan2($s3$, $c3$), B: $\left[\text{ratsimp}\left(\sqrt{x^2+y^2}\right), \text{ratsimp}(z-L1)\right]$, A: [trigsimp(ratsimp(cos $(q3)\,L3+L2$)), |trigsimp(ratsimp(sin $(q3)\,L3$))|], $c2$: $[A_1\,B_1+A_2\,B_2, (-A_1)\,B_1+A_2\,B_2, A_1\,B_1-A_2\,B_2, (-A_1)\,B_1-A_2\,B_2]$, $s2$: $[(-A_2)\,B_1+A_1\,B_2, A_2\,B_1+A_1\,B_2, A_2\,B_1+A_1\,B_2, (-A_2)\,B_1+A_1\,B_2]$, $q2$: [atan2(ratsimp($s2_1$), ratsimp($c2_1$)), atan2(ratsimp($s2_2$), ratsimp($c2_2$)), atan2(ratsimp($s2_3$), ratsimp($c2_3$)), atan2(ratsimp($s2_4$), ratsimp($c2_4$))], den: $[L3\cos(q3+q2_1)+L2\cos(q2_1), L3\cos(q3+q2_2)+L2\cos(q2_2), L3\cos(-q3+q2_3)+L2\cos(q2_3), L3\cos(-q3+q2_4)+L2\cos(q2_4)]$, $c1$: $\left[\text{ratsimp}\left(\frac{x}{\text{den}_1}\right), \text{ratsimp}\left(\frac{x}{\text{den}_2}\right), \text{ratsimp}\left(\frac{x}{\text{den}_3}\right), \text{ratsimp}\left(\frac{x}{\text{den}_4}\right)\right]$, $s1$: $\left[\text{ratsimp}\left(\frac{y}{\text{den}_1}\right), \text{ratsimp}\left(\frac{y}{\text{den}_2}\right), \text{ratsimp}\left(\frac{y}{\text{den}_3}\right), \text{ratsimp}\left(\frac{y}{\text{den}_4}\right)\right]$, $q1$: [atan2($s1_1$, $c1_1$), atan2($s1_2$, $c1_2$), atan2($s1_3$, $c1_3$), atan2($s1_4$, $c1_4$)], res: $[[q1_1, q2_1, q3], [q1_2, q2_2, q3], [q1_3, q2_3, -q3], [q1_4, q2_4, -q3]]$)

(%i21)
```
orientation(Qdiretta):=block([sx,cx,sy,cy,phiy1,phiy2,phiz1,phiz2,phix1,
phix2,sz,sxfirst,second,res],

                              rotation:isRotation(Qdiretta),
                              if rotation=1 then(
                              sy:Qdiretta[3][1],

                              if sy=1 or sy=-1 then print("soluzione
singolare")
                              else(
                              cy:sqrt(1-sy^2),
                              phiy1:atan2(-sy,cy),
                              phiy2:atan2(-sy,-cy),

                              sx:Qdiretta[3][2]/cy,
                              cx:Qdiretta[3][3]/cy,
                              phix1:atan2(sx,cx),
                              phix2:atan2(-sx,cx),
                              cz:Qdiretta[1][1]/cy,
                              sz:Qdiretta[2][1]/cy,
                              phiz1:atan2(+sz,cz),
                              phiz2:atan2(-sz,cz),
                              first:[phix1,phiy1,phiz1],
                              second:[phix2,phiy2,phiz2],

                              res:[first,second])
                              )


);
```

(%o21) orientation(Qdiretta) := **block** ([sx, cx, sy, cy, phiy1, phiy2, phiz1, phiz2, phix1, phix2, sz, sxfirst, second, res], rotation: isRotation(Qdiretta), **if** rotation = 1 **then** (sy: $(\text{Qdiretta}_3)_1$, **if** sy $=1 \vee$ sy $=-1$ **then** print(soluzione singolare) **else** (cy: $\sqrt{1-\text{sy}^2}$, phiy1: atan2(−sy, cy), phiy2:

atan2(−sy, −cy), sx: $\frac{(\text{Qdiretta}_3)_2}{\text{cy}}$, cx: $\frac{(\text{Qdiretta}_3)_3}{\text{cy}}$, phix1: atan2(sx, cx), phix2: atan2(−sx, cx), cz: $\frac{(\text{Qdiretta}_1)_1}{\text{cy}}$, sz: $\frac{(\text{Qdiretta}_2)_1}{\text{cy}}$, phiz1: atan2(+sz, cz), phiz2: atan2(−sz, cz), first: [phix1, phiy1, phiz1], second: [phix2, phiy2, phiz2], res: [first, second])))

```
(%i22) invAntropomorfo(x,y,z,L1,L2,L3,alpha,beta,gamma):=block(
       [pos,R,orient],
        pos:calculate(x,y,z,L1,L2,L3),
        R:rodriguez([0,0,1],alpha).
          rodriguez([0,1,0],beta).
          rodriguez([1,0,0],gamma),
        print("Rzyx=",R),
        orient:orientation(R),
        print("Position=",pos),
        print("Orientation=",orient)
        )
```

(%o22) invAntropomorfo$(x, y, z, L1, L2, L3, \alpha, \beta, \gamma)$:= **block** ([pos, R, orient], pos: calculate$(x, y, z, L1, L2, L3)$, R: rodriguez$([0, 0, 1], \alpha) \cdot$ rodriguez$([0, 1, 0], \beta) \cdot$ rodriguez$([1, 0, 0], \gamma)$, print(Rzyx= , R), orient: orientation(R), print(Position= , pos), print(Orientation= , orient))

```
(%i23) invAntropomorfo(1,1,1,1,1,2,%pi/2,3*%pi/4,%pi/4);
```

$$\text{Rzyx=}\begin{pmatrix} 0 & -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ -\frac{1}{\sqrt{2}} & \frac{1}{2} & \frac{1}{2} \\ -\frac{1}{\sqrt{2}} & -\frac{1}{2} & -\frac{1}{2} \end{pmatrix}$$

$$\text{Position=}\left[\left[\frac{\pi}{4}, \arctan\left(\sqrt{7}\right)-\pi, \pi-\arctan\left(\frac{\sqrt{7}}{3}\right)\right], \left[-\frac{3\pi}{4}, \arctan\left(\sqrt{7}\right), \pi-\arctan\left(\frac{\sqrt{7}}{3}\right)\right], \left[\frac{\pi}{4}, \pi-\arctan\left(\sqrt{7}\right), \arctan\left(\frac{\sqrt{7}}{3}\right)-\pi\right], \left[-\frac{3\pi}{4}, -\arctan\left(\sqrt{7}\right), \arctan\left(\frac{\sqrt{7}}{3}\right)-\pi\right]\right]$$

$$\text{Orientation=}\left[\left[-\frac{3\pi}{4}, \frac{\pi}{4}, -\frac{\pi}{2}\right], \left[\frac{3\pi}{4}, \frac{3\pi}{4}, \frac{\pi}{2}\right]\right]$$

(%o23) $\left[\left[-\frac{3\pi}{4}, \frac{\pi}{4}, -\frac{\pi}{2}\right], \left[\frac{3\pi}{4}, \frac{3\pi}{4}, \frac{\pi}{2}\right]\right]$

Singolarità

```
Maxima 5.44.0 http://maxima.sourceforge.net
using Lisp SBCL 2.0.0
Distributed under the GNU Public License. See the file COPYING.
Dedicated to the memory of William Schelter.
The function bug_report() provides bug reporting information.
(%i1) x:cos (q[1])*(L[3]*cos (q[3]+q[2])+L[2]*cos (q[2]));
```

(%o1) $\cos(q_1)\left(L_3\cos(q_3+q_2)+L_2\cos(q_2)\right)$

```
(%i2) y:sin (q[1])*(L[3]*cos (q[3]+q[2])+L[2]*cos (q[2]));
```

(%o2) $\sin(q_1)\left(L_3\cos(q_3+q_2)+L_2\cos(q_2)\right)$

```
(%i3) z:L[3]*sin (q[3]+q[2])+L[2]*sin (q[2])+L[1];
```

(%o3) $L_3\sin(q_3+q_2)+L_2\sin(q_2)+L_1$

(%i4)
```
J:J:matrix([diff(x,q[1]),diff(x,q[2]),diff(x,q[3])],
           [diff(y,q[1]),diff(y,q[2]),diff(y,q[3])],
           [diff(z,q[1]),diff(z,q[2]),diff(z,q[3])]);
```

(%o4) $(-\sin(q_1)(L_3\cos(q_3+q_2)+L_2\cos(q_2)), \cos(q_1)(-L_3\sin(q_3+q_2)-L_2\sin(q_2)), -L_3\cos(q_1)\sin(q_3+q_2); \cos(q_1)(L_3\cos(q_3+q_2)+L_2\cos(q_2)), \sin(q_1)(-L_3\sin(q_3+q_2)-L_2\sin(q_2)), -L_3\sin(q_1)\sin(q_3+q_2); 0, L_3\cos(q_3+q_2)+L_2\cos(q_2), L_3\cos(q_3+q_2))$

(%i5)
```
dJ:factor(trigsimp(determinant(J)));
```

(%o5) $-L_2L_3(L_3\cos(q_3+q_2)+L_2\cos(q_2))(\cos(q_2)\sin(q_3+q_2)-\sin(q_2)\cos(q_3+q_2))$

$$L_3\sin(q_2)\sin(q_3)-L_3\cos(q_2)\cos(q_3)=L_2\cos(q_2)$$

$$\begin{cases} b=L_3\sin(q_2) \\ a=L_3\cos(q_2) \end{cases}$$

$$b\sin(q_3)-a\cos(q_3)=\frac{L_2}{L_3}a$$

$$b\sin(q_3)=a\left(\cos(q_3)+\frac{L_2}{L_3}\right)$$

$$\mathbf{q_3\neq 0}\rightarrow b=a\left(\frac{L_3\cos(q_3)+L_2}{L_3\sin(q_3)}\right)$$

$$a^2+a^2\left(\frac{L_3\cos(q_3)+L_2}{L_3\sin(q_3)}\right)^2=L_3^2$$

$$a^2\left(\frac{L_3^2\sin(q_3^2)+(L_3\cos(q_3)+L_2)^2}{L_3^2\sin(q_3^2)}\right)=L_3^2$$

$$a^2\left(\frac{L_3^2+L_2^2+2L_3L_2\cos(q_3)}{L_3^2\sin(q_3)^2}\right)=L_3^2$$

$$a^2=\frac{L_3^4\sin(q_3^2)}{L_3^2+L_2^2+2L_3L_2\cos(q_3)}\rightarrow a=\pm\frac{L_3^2\sin(q_3)}{\sqrt{L_3^2+L_2^2+2L_3L_2\cos(q_3)}}$$

$$b=\frac{\pm L_3^2\sin(q_3)}{\sqrt{L_3^2+L_2^2+2L_3L_2\cos(q_3)}}\left(\frac{L_3\cos(q_3)+L_2}{L_3\sin(q_3)}\right)=\frac{\pm L_3(L_3\cos(q_3)+L_2)}{\sqrt{2L_3^2+L_2^2+2L_3L_2\cos(q_3)}}$$

$$\begin{cases} \pm\frac{L_3(L_3\cos(q_3)+L_2)}{\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2}}=\sin(q_2) \\ \pm\frac{L_3^2\sin(q_3)}{\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2}}=\cos(q_2) \end{cases}$$

$$q_2=\mathrm{atan2}(\pm(L_3\cos(q_3)+L_2),\pm\sin(q_3))$$

$$\mathbf{q_3=0}\rightarrow a\left(1+\frac{L_2}{L_3}\right)=0\rightarrow a=0$$

$$\begin{cases} L_3\sin(q_2)=b \\ L_3\cos(q_2)=0 \end{cases}\rightarrow\begin{cases} \cos(q_2)=0 \\ \sin(q_2)=\frac{b}{L_3} \end{cases}\rightarrow q_2=\frac{\pi}{2}, b=L_3$$

Riassumendo:

$$q_3\neq 0\wedge q_2=\mathrm{atan2}(\pm L_3(L_3\cos(q_3)+L_2),\pm L_3^2\sin(q_3))\vee q_3=0\wedge q_2=\frac{\pi}{2}$$

Verifica che le soluzioni trovate annullino il determinante:

```
(%i6) subst([q[3]=0,q[2]=%pi/2],dJ);
```

(%o6) 0

```
(%i7) trigsimp(trigexpand(subst([q[2]=atan2((L[3]*(L[3]*cos(q[3])+L[2])),
      L[3]^2*sin(q[3]))],dJ)));
```

(%o7) 0

Caso $q_3=0\wedge q_2=\frac{\pi}{2}$:

```
(%i8) Jq32:subst([q[3]=0,q[2]=%pi/2],J);
```

(%o8) $$\begin{pmatrix} 0 & (-L_3-L_2)\cos(q_1) & -L_3\cos(q_1) \\ 0 & (-L_3-L_2)\sin(q_1) & -L_3\sin(q_1) \\ 0 & 0 & 0 \end{pmatrix}$$

```
(%i9) nullspace(Jq32);
```

Proviso: notequal$(-L_3\cos(q_1),0)$

(%o9) $$\mathrm{span}\left(\begin{pmatrix} 0 \\ -L_3\cos(q_1) \\ (L_3+L_2)\cos(q_1) \end{pmatrix},\begin{pmatrix} -L_3\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right)$$

Se $q_1\neq 0$, le singolarità di velocità si hanno per $v\in \mathrm{Im}\left\{\begin{pmatrix} 0 \\ -L_3\cos(q_1) \\ (L_3+L_2)\cos(q_1) \end{pmatrix},\begin{pmatrix} -L_3\cos(q_1) \\ 0 \\ 0 \end{pmatrix}\right\}$.

Se $q_1=0$:

```
(%i11) Jq321:subst(q[1]=0,Jq32);
```

(%o11) $$\begin{pmatrix} 0 & -L_3-L_2 & -L_3 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

```
(%i12) nullspace(Jq321);
```

Proviso: notequal$(-L_3,0)$

(%o12) $$\mathrm{span}\left(\begin{pmatrix} 0 \\ -L_3 \\ L_3+L_2 \end{pmatrix},\begin{pmatrix} -L_3 \\ 0 \\ 0 \end{pmatrix}\right)$$

le singolarità di velocità si hanno per $v\in \mathrm{Im}\left\{\begin{pmatrix} 0 \\ -L_3 \\ L_3+L_2 \end{pmatrix},\begin{pmatrix} -L_3 \\ 0 \\ 0 \end{pmatrix}\right\}$

Caso $q_3\neq 0\wedge q_2=\mathrm{atan2}(\pm L_3(L_3\cos(q_3)+L_2),\pm L_3^2\sin(q_3))$:

```
(%i14) Jq32:trigsimp(trigexpand(subst([q[2]=atan2((L[3]*(L[3]*cos(q[3])+L[2])),
       L[3]^2*sin(q[3]))],J)));
```

(%o14) $$\left(0,-\frac{2L_2L_3^2\cos(q_1)\cos(q_3)+(L_3^3+L_2^2L_3)\cos(q_1)}{\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2}},\right.$$
$$-\frac{\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2}\,(L_2\cos(q_1)\cos(q_3)+L_3\cos(q_1))}{2L_2L_3\cos(q_3)+L_3^2+L_2^2};0,-\frac{2L_2L_3^2\sin(q_1)\cos(q_3)+(L_3^3+L_2^2L_3)\sin(q_1)}{\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2}},$$
$$\left.-\frac{\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2}\,(L_2\sin(q_1)\cos(q_3)+L_3\sin(q_1))}{2L_2L_3\cos(q_3)+L_3^2+L_2^2};0,0,-\frac{L_2\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2}\sin(q_3)}{2L_2L_3\cos(q_3)+L_3^2+L_2^2}\right)$$

```
(%i15) nullspace(Jq32);
```

Proviso: notequal$\left(-\cos(q_1)\sqrt{2L_2L_3^3\cos(q_3)+L_3^4+L_2^2L_3^2},0\right)\wedge$
notequal$((2L_2^2L_3^3\cos(q_1)\cos(q_3)+(L_2L_3^4+L_2^3L_3^2)\cos(q_1))\sin(q_3),0)$

(%o15) $\operatorname{span}\left(\begin{pmatrix} (2\,L_2^2\,L_3^3\cos(q_1)\cos(q_3)+(L_2\,L_3^4+L_2^3\,L_3^2)\cos(q_1))\sin(q_3) \\ 0 \\ 0 \end{pmatrix}\right)$

Se $q_1 \neq 0$, si hanno singolarità di velocita per $v \in \operatorname{Im}\left\{\begin{pmatrix} (2\,L_2^2\,L_3^3\cos(q_1)\cos(q_3)+(L_2\,L_3^4+L_2^3\,L_3^2)\cos(q_1))\sin(q_3) \\ 0 \\ 0 \end{pmatrix}\right\}$.

Se $q_1 = 0$:

```
(%i16) Jq321:subst(q[1]=0,Jq32);
```

(%o16) $$\begin{pmatrix} 0 & -\frac{2\,L_2\,L_3^2\cos(q_3)+L_3^3+L_2^2\,L_3}{\sqrt{2\,L_2\,L_3^3\cos(q_3)+L_3^4+L_2^2\,L_3^2}} & -\frac{(L_2\cos(q_3)+L_3)\sqrt{2\,L_2\,L_3^3\cos(q_3)+L_3^4+L_2^2\,L_3^2}}{2\,L_2\,L_3\cos(q_3)+L_3^2+L_2^2} \\ 0 & 0 & 0 \\ 0 & 0 & -\frac{L_2\sqrt{2\,L_2\,L_3^3\cos(q_3)+L_3^4+L_2^2\,L_3^2}\sin(q_3)}{2\,L_2\,L_3\cos(q_3)+L_3^2+L_2^2} \end{pmatrix}$$

```
(%i17) nullspace(Jq321);
```

Proviso: $\operatorname{notequal}\left(-\sqrt{2\,L_2\,L_3^3\cos(q_3)+L_3^4+L_2^2\,L_3^2}, 0\right) \wedge \operatorname{notequal}((2\,L_2^2\,L_3^3\cos(q_3)+L_2\,L_3^4+L_2^3\,L_3^2)\sin(q_3), 0)$

(%o17) $\operatorname{span}\left(\begin{pmatrix} (2\,L_2^2\,L_3^3\cos(q_3)+L_2\,L_3^4+L_2^3\,L_3^2)\sin(q_3) \\ 0 \\ 0 \end{pmatrix}\right)$

(%i18)