Calcolare $Q_{i-1,i}$

La matrice di trasformazione omogenea $Q_{i-1,i}$ consente di sovrappore il sistema di riferimento R_{i-1} con il sistema di riferimento R_i tramite 2 matrici di avvitamento $A_v(z,\theta,d), A_v(x,\alpha,a)$ nel seguente ordine.

Per ogni grado di libertà:

$$Q_{i-1,i} = A_v(z,\theta_i,d_i)\,A_v(x,\alpha_i,a_i)$$

```
(%i1) inverseLaplace(SI,theta):=block([res],
                         M:SI,
                         MC:SI,
                         for i:1 thru 3 do(
                           for j:1 thru 3 do
                              (
                                aC:M[i,j],
                                b:ilt(aC,s,theta),
                                MC[i,j]:b
                               )
                           ),
                         res:MC

                      )
```

(%o1) inverseLaplace(SI, ϑ) := **block** ([res], M: SI, MC: SI, **for** i **thru** 3 **do for** j **thru** 3 **do** (aC: $M_{i,j}$, b: ilt(aC, s, ϑ), MC$_{i,j}$: b), res: MC)

```
(%i2) rotLaplace(k,theta):=block([res],
                      S:ident(3),
                      I:ident(3),
                    for i:1 thru 3 do
                      (
                      for j:1 thru 3 do
                         (
                          if i=j
                             then S[i][j]:0
                          elseif j>i
                             then (
                            temp:(-1)^(j-i)*k[3-remainder(i+j,3)],
                                   S[i][j]:temp,
                                   S[j][i]:-temp

                                    )
                          )
                      ),
                      res:inverseLaplace(invert(s*I-S),theta)


                    )
```

(%o2) rotLaplace(k,ϑ) := **block** ([res], S: ident(3), I: ident(3), **for** i **thru** 3 **do for** j **thru** 3 **do if** $i=j$ **then** $(S_i)_j$: 0 **elseif** $j>i$ **then** (temp: $(-1)^{j-i}\,k_{3-\mathrm{remainder}(i+j,3)}$, $(S_i)_j$: temp, $(S_j)_i$: $-$temp), res: inverseLaplace(invert($s\,I-S$), ϑ))

```
(%i3) Av(v,theta,d):=block([res],
                        Trot:rotLaplace(v,theta),
                        row:matrix([0,0,0,1]),
                        Atemp:addcol(Trot,d*transpose(v)),
                        A:addrow(Atemp,row),
                        res:trigsimp(trigrat(trigreduce(trigexpand(A))))
                        )
```

(%o3) $\mathrm{Av}(v, \vartheta, d) := \textbf{block}\ ([\mathrm{res}], \mathrm{Trot}\colon \mathrm{rotLaplace}(v, \vartheta), \mathrm{row}\colon \begin{pmatrix} 0 & 0 & 0 & 1 \end{pmatrix}, \mathrm{Atemp}\colon \mathrm{addcol}(\mathrm{Trot}, d\,\mathrm{transpose}(v)), A\colon \mathrm{addrow}(\mathrm{Atemp}, \mathrm{row}), \mathrm{res}\colon \mathrm{trigsimp}(\mathrm{trigrat}(\mathrm{trigreduce}(\mathrm{trigexpand}(A)))))$

```
(%i4) Az(theta,d):=Av([0,0,1],theta,d);
```

(%o4) $\mathrm{Az}(\vartheta, d) := \mathrm{Av}([0, 0, 1], \vartheta, d)$

```
(%i5) Az(theta,d);
```

(%o5) $$\begin{pmatrix} \cos(\vartheta) & -\sin(\vartheta) & 0 & 0 \\ \sin(\vartheta) & \cos(\vartheta) & 0 & 0 \\ 0 & 0 & 1 & d \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(%i6) Ax(alpha,a):=Av([1,0,0],alpha,a);
```

(%o6) $\mathrm{Ax}(\alpha, a) := \mathrm{Av}([1, 0, 0], \alpha, a)$

```
(%i7) Ax(alpha,a);
```

(%o7) $$\begin{pmatrix} 1 & 0 & 0 & a \\ 0 & \cos(\alpha) & -\sin(\alpha) & 0 \\ 0 & \sin(\alpha) & \cos(\alpha) & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

La matrice Q riceve in input ϑ,d,α,a:
-ϑ,α sono angoli corrispondenti alle rotazioni necessarie a far sovrapporre il sistema di riferimento R_{i-1} con il sistema di riferimento R_i rispettivamente con la matrice di avvitamento $A_v(z, \theta, d)$ e $A_v(x, \alpha, a)$;
-d,a sono posizioni che corrispondo alle traslazioni necessarie a far sovrapporre il sistema di riferimento R_{i-1} con il sistema di riferimento R_i rispettivamente lungo l'asse z e x corrispondenti alle matrici di avvitamento $A_v(z, \theta, d)$ e $A_v(x, \alpha, a)$;

```
(%i8) Q(theta,d,alpha,a):=block([res],
      res:trigexpand(trigsimp(trigrat(trigreduce(trigexpand(Az(theta,d).Ax(alpha,
      a))))))
                                         )
```

(%o8) $Q(\vartheta, d, \alpha, a) := \textbf{block}\ ([\mathrm{res}], \mathrm{res}\colon \mathrm{trigexpand}(\mathrm{trigsimp}(\mathrm{trigrat}(\mathrm{trigreduce}(\mathrm{trigexpand}(\mathrm{Az}(\vartheta, d) \cdot \mathrm{Ax}(\alpha, a)))))))$

```
(%i9) Q(theta,d,alpha,a)
```

(%o9) $$\begin{pmatrix} \cos(\vartheta) & -\cos(\alpha)\sin(\vartheta) & \sin(\alpha)\sin(\vartheta) & a\cos(\vartheta) \\ \sin(\vartheta) & \cos(\alpha)\cos(\vartheta) & -\sin(\alpha)\cos(\vartheta) & a\sin(\vartheta) \\ 0 & \sin(\alpha) & \cos(\alpha) & d \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(%i10)
(%i13)
```