Asse, angolo, spostamento - Matrice di avvitamento

Scrivere una procedura che, dati asse, angolo, spostamento, generi la corrispondente matrice di avvitamento

Una matrice di avvitamento è una matrice ottenuta dalla rotazione e traslazione lungo uno stesso asse. Questa matrice ha la seguente struttura:

$$A_v(\theta, d) = \begin{pmatrix} e^{S(v)\theta} & d\,v \\ 0 & 1 \end{pmatrix}$$

In particolare:

- $\vartheta = 0 \Longrightarrow$traslazione;
- $d = 0 \Longrightarrow$ rotazione;

Poiché la traslazione e la rotazione sono effettuate lungo lo stesso asse, commutano. Quindi, una roto-traslazione è equivalente ad una trasla-rotazione.

$$T_R T_T \equiv T_T T_R$$

$$T_R = \begin{pmatrix} e^{S(v)\theta} & 0 \\ 0 & 1 \end{pmatrix}, T_T = \begin{pmatrix} I & d\,v \\ 0 & 1 \end{pmatrix},$$

$$(1) T_R T_T = \begin{pmatrix} e^{S(v)\theta} & 0 \\ 0 & 1 \end{pmatrix} \begin{pmatrix} I & d\,v \\ 0 & 1 \end{pmatrix} = \begin{pmatrix} e^{S(v)\theta} & d\,e^{S(v)\theta} v \\ 0 & 1 \end{pmatrix}$$

$$(2) T_T T_R = \begin{pmatrix} I & d\,v \\ 0 & 1 \end{pmatrix} \begin{pmatrix} e^{S(v)\theta} & 0 \\ 0 & 1 \end{pmatrix} = \begin{pmatrix} e^{S(v)\theta} & d\,v \\ 0 & 1 \end{pmatrix}$$

I prodotti matriciali (1) e (2) sono identici tenendo conto che v è l'asse di rotazione corrispondente alla rotazione effettuata:

$$e^{S(v)\theta} v = v$$

poiché v è l'asse di rotazione.

La funzione inverseLaplace(SI) calcola e restituisce in output l'antitrasformata di Laplace scorrendo tutti gli elementi della matrice data in input.

```
(%i1) inverseLaplace(SI,theta):=block([res],
                                      M:SI,
                                      MC:SI,
                                      for i:1 thru 3 do(
                                        for j:1 thru 3 do
                                            (
                                              aC:M[i,j],
                                              b:ilt(aC,s,theta),
                                              MC[i,j]:b
                                             )
                                          ),
                                      res:MC

                                 )
```

(%o1) inverseLaplace(SI, ϑ) := **block** ([res], M: SI, MC: SI, **for** i **thru** 3 **do for** j **thru** 3 **do** (aC: $M_{i,j}$, b: ilt(aC, s, ϑ), MC$_{i,j}$: b), res: MC)

La funzione rotLaplace(k,ϑ) riceve in input un vettore v e ϑ per calcolare la matrice di rotazione relativa all'asse di rotazione v e angolo ϑ. In seguito, invoca la funzione inverseLaplace per effettuare l'inversa di laplace e, quindi, restituire in output l'effettiva matrice di rotazione.

```
(%i2) rotLaplace(k,theta):=block([res],
                            S:ident(3),
                            I:ident(3),
                          for i:1 thru 3 do
                            (
                            for j:1 thru 3 do
                               (
                                 if i=j
                                    then S[i][j]:0
                                 elseif j>i
                                    then (
                                   temp:(-1)^(j-i)*k[3-remainder(i+j,3)],
                                          S[i][j]:temp,
                                          S[j][i]:-temp

                                           )
                                 )
                             ),
                            res:inverseLaplace(invert(s*I-S),theta)


                          )
```

(%o2) $\mathrm{rotLaplace}(k,\vartheta) := \textbf{block}\ ([\mathrm{res}], S\colon \mathrm{ident}(3), I\colon \mathrm{ident}(3),$
$\textbf{for}\ i\ \textbf{thru}\ 3\ \textbf{do}\ \textbf{for}\ j\ \textbf{thru}\ 3\ \textbf{do}\ \textbf{if}\ i=j\ \textbf{then}\ (S_i)_j\colon 0\ \textbf{elseif}\ j>i\ \textbf{then}\ (\mathrm{temp}\colon$
$(-1)^{j-i}\, k_{3-\mathrm{remainder}(i+j,3)}, (S_i)_j\colon \mathrm{temp}, (S_j)_i\colon -\mathrm{temp}), \mathrm{res}\colon \mathrm{inverseLaplace}(\mathrm{invert}(s\,I-S),\vartheta))$

La funzione $A_v(v,\theta,d)$ riceve in input l'asse di rotazione v, l'angolo ϑ e la traslazione d e restituisce la corrispondente matrice di avvitamento A_v:

$$A_v(\theta,d) = \begin{pmatrix} e^{S(v)\theta} & d\,v \\ 0 & 1 \end{pmatrix}$$

```
(%i3) Av(v,theta,d):=block([res],
                            Trot:rotLaplace(v,theta),
                            row:matrix([0,0,0,1]),
                            Atemp:addcol(Trot,d*transpose(v)),
                            A:addrow(Atemp,row),
                            res:trigsimp(trigrat(trigreduce(trigexpand(A))))
                            )
```

(%o3) $\mathrm{Av}(v,\vartheta,d) := \textbf{block}\ ([\mathrm{res}], \mathrm{Trot}\colon \mathrm{rotLaplace}(v,\vartheta), \mathrm{row}\colon \begin{pmatrix} 0 & 0 & 0 & 1 \end{pmatrix}, \mathrm{Atemp}\colon \mathrm{addcol}(\mathrm{Trot},$
$d\,\mathrm{transpose}(v)), A\colon \mathrm{addrow}(\mathrm{Atemp},\mathrm{row}), \mathrm{res}\colon \mathrm{trigsimp}(\mathrm{trigrat}(\mathrm{trigreduce}(\mathrm{trigexpand}(A)))))$

```
(%i4) A[z](theta,d):=Av([0,0,1],theta,d);
```

(%o4) $A_z(\vartheta,d) := \mathrm{Av}([0,0,1],\vartheta,d)$

```
(%i5) A[z](theta,d)
```

(%o5)
$$\begin{pmatrix} \cos(\vartheta) & -\sin(\vartheta) & 0 & 0 \\ \sin(\vartheta) & \cos(\vartheta) & 0 & 0 \\ 0 & 0 & 1 & d \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

(%i6) `A[x](alpha,a):=Av([1,0,0],alpha,a);`

(%o6) $A_x(\alpha, a) := \mathrm{Av}([1,0,0], \alpha, a)$

(%i7) `A[x](alpha,a)`

(%o7)
$$\begin{pmatrix} 1 & 0 & 0 & a \\ 0 & \cos(\alpha) & -\sin(\alpha) & 0 \\ 0 & \sin(\alpha) & \cos(\alpha) & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

(%i8)